
Chapter 1

Introduction

In this thesis, I show that it is possible to reduce the problem of checking
strong reactive bisimilarity, as introduced by Rob van Glabbeek in [vG20],
to checking ordinary strong bisimilarity. I do this by specifying a mapping
that yields a model of the closed system consisting of some original (reactive)
system and its environment. I formalised all concepts discussed in this thesis,
and conducted all the proofs, in the interactive proof assistant Isabelle.
Reactive systems are systems that continuously interact with their envi-
ronment (e.g. a user) and whose behaviour is largely dependent on this
interaction [HP85]. They can be modelled using labelled transition systems
(LTSs) [Kel76]; roughly, an LTS is a labelled directed graph, whose nodes
correspond to states of a reactive system and whose edges correspond to
transitions between those states.1

A user interacting with some system can only perceive it in terms of the
interactions it reacts to, i.e. the internal state of the system is hidden from
the user. This begets a notion of behavioural/observational equivalence: two
non-identical systems may exhibit equivalent behaviour as observed by the
user. The simplest such equivalence is known as strong bisimilarity.
In classical LTSs, a system cannot react to the absence of interaction, as it
would be assumed to simply wait for any interaction. Intuitively, however, a
system may be equipped with a clock and perform some activity when it has
seen no interaction from the user for a specified time. Such a system would
not be describable with classical LTS semantics. Amongst these systems
are, e.g., systems implementing mutual exclusion protocols [vG20].

1The topics of this thesis are applicable to any such graphs in an abstract way. However,
I will continue to use motivations and terminology derived from the interpretation of LTSs
as reactive systems.

6 CHAPTER 1. INTRODUCTION

In [vG21], Rob van Glabbeek introduces labelled transition systems with
time-outs (LTSt)2, which allow for modelling such systems as well. The
appertaining equivalence is given in [vG20] as strong reactive bisimilarity.
For the first main result of this thesis, I show that it is possible to reduce
checking strong reactive bisimilarity to checking strong bisimilarity. This is
in line with reductions of other behavioural equivalences to strong bisimil-
arity. For example, a strategy used to reduce weak bisimilarity to strong
bisimilarity is called saturation and is described in [AIS11, section 3.2.5].
The strategy used for reducing reactive bisimilarity to strong bisimilarity is
based on the fact that the concept of strong reactive bisimilarity requires an
explicit consideration of the environments in which specified systems may
exist. Concretely, I specify a mapping from LTSts to LTSs, where each state
of the mapped LTS corresponds to a state of the original LTSt in some
specific environment. This approach has been hinted at by van Glabbeek at
various points in [vG20], but has not been made explicit.
The reduction of reactive bisimilarity could be of use in the context of auto-
mated model checking tools: there are known algorithms for checking equi-
valences (e.g. see [AIS11]) and tools with efficient implementations thereof;3
instead of implementing an algorithm for checking strong reactive bisim-
ilarity from scratch, an implementation of the reduction would allow the
use of these existing implementations. Moreover, the mapping used for the
reduction may aid in the analysis of system specifications using LTSts, by
providing a more explicit view at the system.
Another interesting way to examine the behaviour of an LTS is through
the use of modal logics, where formulas describe certain properties and are
evaluated on states of an LTS. A commonly used modal logic is known
as Hennessy-Milner logic (HML). An extension of HML for evaluation on
states of an LTSt is also given in [vG20]; I will refer to this extension as
Hennessy-Milner logic with time-outs (HMLt).
For the second main result of this thesis, I show that it is possible to reduce
formula satisfaction of HMLt on LTSts to formula satisfaction of HML on
LTSs (using another mapping for formulas, along with the mapping from
the first reduction).

2He does not use that specific term or abbreviation, however.
3e.g. see LTSmin at github.com/utwente-fmt/ltsmin

https://github.com/utwente-fmt/ltsmin

How This Thesis is Structured / Isabelle

The remainder of this thesis is split into Foundations (chapter 2), where
LTSs, bisimilarity, and Hennessy-Milner logic, all without and with time-
outs, are discussed and formalised, and The Reductions (chapter 3), where
the reduction of bisimilarity and the reduction of formula satisfaction are
presented in detail and proved.
All the main topics of this thesis have been formalised, and all proofs con-
ducted, using the interactive proof assistant Isabelle. More information on
Isabelle and a short introduction to the most important concepts can be
found in appendix A.
This thesis document itself was generated using the Isabelle document pre-
paration system (see [Wen21a]), which generates LATEXmarkup from Isabelle
code (and, of course, integrates markup written manually). This allowed me
to integrate all the Isabelle code directly into the thesis document. How-
ever, almost all proofs are hidden (and replaced simply by 〈proof 〉) and some
lemmas excluded. In these cases, an indication of the proof strategy used is
given in text. A web version of this thesis, that includes all formalisations,
propositions, and proofs, as well as all the text, can be found on GitHub
Pages, with one page for each section of this thesis.4

All of the sections of chapters 2 and 3 are split into two parts: one contain-
ing a prosaic and mathematical description of the topics, and one containing
the (documented) formalisation/implementation in Isabelle. I try to clearly
distinguish between mathematical structures and their implementation. Al-
though the two are, necessarily, closely related, they are not identical. The
former is written in LATEXmath mode in this italic font, the latter is Isabelle
code in this monospaced font.

4see maxpohlmann.github.io/Reducing-Reactive-to-Strong-Bisimilarity

https://maxpohlmann.github.io/Reducing-Reactive-to-Strong-Bisimilarity


Chapter 2

Foundations

In this chapter, the concepts that are relevant for the main part of this thesis
will be introduced in text, as well as formalised in Isabelle. The formalisa-
tions of sections 2.1 and 2.2 are based on those done by Benjamin Bisping
in [BN19] (the code is available on GitHub1). All other formalisations were
done as part of this thesis.

2.1 Labelled Transition Systems

A Labelled Transition System (LTS) consists of a set of processes (or states)
Proc, a set of actions Act , and a relation of transitions · ·−→ · ⊆ Proc×Act ×
Proc which directedly connect two processes by an action (the action being
the label of the transition) [AILS07]. We call a transition labelled by an
action α an α-transition.
LTSs can model reactive systems, as discussed in chapter 1. A process of
an LTS, then, corresponds to a momentary state of a reactive system. The
outgoing transitions of each process correspond to the actions the reactive
system could perform in that state (yielding a subsequent process/state),
if facilitated by the environment. The choice between the facilitated trans-
itions of a process is non-deterministic.
This facilitation can be thought of as a set of actions that the environment
allows in a given moment, or, more intuitively, as a set of simultaneous
inputs from the environment to which the system may react. We call the
actions not allowed by the environment in a given moment blocked.
The environment can also observe which transitions a system performs and
react by changing the set of allowed actions in response.

1see coupledsim.bbisping.de/code/

8

2.1. LABELLED TRANSITION SYSTEMS 9

In classical treatments of LTSs, the actions that the environment allows,
blocks, or reacts to, are often considered only implicitly. The reason we put
this emphasis on the environment here will become apparent in section 2.4.

Example The process p can perform any of the a-transitions in envir-
onments allowing a and the b-transition in environments allowing b. All
derivative (subsequent) states cannot perform any transition.

p

p1
p2

p3

a
a

b

A hidden action, denoted by τ , allows for additional semantics: a τ -transition
can be performed by a process without any interaction from the environ-
ment. Depending on the specific semantic context, the performance of a
hidden action may also be unobservable (hence the name), i.e. not part of
the observable behaviour of the system.

Some Definitions

The α-derivatives of a state are those states that can be reached with one
α-transition:

Der(p, α) = {p′ | p α−→ p′}.

An LTS is image-finite iff all derivative sets are finite, i.e.:

∀p ∈ Proc, α ∈ Act . Der(p, α) is finite.

Similarly, we can say that an LTS is image-countable iff all derivative sets
are countable, i.e.:

∀p ∈ Proc, α ∈ Act . Der(p, α) is countable.

Note on Metavariable usage

States of an LTS range over p, q, p′, q′, . . . , where p and p′ are used for states
connected by some transition (i.e. p α−→ p′), whereas p and q are used for
states possibly related by some equivalence (e.g. p↔ q), as will be discussed
in the next section.
An arbitrary action of an LTS will be referenced by α, whereas an arbitrary
visible (i.e. non-hidden) action will be referenced by a.

Practical Example

Let us take a detour away from purely theoretical deliberations and consider
a real-world machine that may be modelled as an LTS. We can imagine a
very simple snack-selling vending machine that accepts only one type of coin
and has individual buttons for each of the snacks. When a coin is inserted
and a button pressed, the machine dispenses the desired snack and is then
ready to accept coins once again. Because the dispensing of the snack itself
requires no interaction from the user, we model it as a τ -transition.

p

p1

p2
p3

p4

coin

choc
nuts

crisps
τ τ

τ

Isabelle

The sets of states and actions are formalised by type variables 's and 'a,
respectively. A specific LTS on these sets is then determined entirely by its
set of transitions, denoted by the predicate tran. We associate it with a
more readable notation (p 7−→α p' for p α−→ p′).
locale lts =

fixes tran :: 〈's ⇒ 'a ⇒ 's ⇒ bool〉

(_ 7−→_ _ [70, 70, 70] 80)
begin

The other definitions can be formalised in a straightforward manner.
abbreviation derivatives :: 〈's ⇒ 'a ⇒ 's set〉

(〈Der'(_, _')〉 [50, 50] 1000)
where 〈Der(p, α) ≡ {p'. p 7−→α p'}〉

definition image_finite :: 〈bool〉

where 〈image_finite ≡ (∀ p α. finite Der(p, α))〉

definition image_countable :: 〈bool〉

where 〈image_countable ≡ (∀ p α. countable Der(p, α))〉

These two properties concern the entire LTS at hand (given by the locale
context) and will be useful when we want to state propositions that only
hold for LTSs that satisfy these properties.
end — of locale lts

We formalise LTSs with hidden actions as an extension of ordinary LTSs
with a fixed action τ .

locale lts_tau = lts tran
for tran :: 〈's ⇒ 'a ⇒ 's ⇒ bool〉 +
fixes τ :: 〈'a〉

2.2 Strong Bisimilarity

As discussed in the previous section, LTSs can describe the behaviour of
reactive systems, and this behaviour is observable by the environment (in
terms of the transitions performed by the system). This begets a notion of
behavioural equivalence, where two processes are said to be behaviourally
equivalent if they exhibit the same (observable) behaviour [AILS07].
Bisimilarity (or strong bisimilarity, to be precise) is the ‘finest extensional
behavioural equivalence […] on processes’ [San12, section 0.1], an extensional
property being one that treats the system in question as a black box, i.e. the
specific state space of the system remains hidden and performed transitions
are only observable in terms of their action-label. This distinguishes bisim-
ilarity from stronger graph equivalences like graph isomorphism, where the
(intensional) identity of processes (graph nodes) is relevant [San11].

Example The processes p and q are strongly bisimilar (written p ↔ q,
following [vG20]), as both can always perform exactly two a-transitions and
no further transitions afterwards. There is no isomorphism between the left
and right subgraphs, as they have a different number of nodes.

p

p1

p2

p3

p4

q

q1

q2

a a

a a

a

a

It is important to note that not only transitions that are performable, but
also those that are not, are relevant.

Example The processes p and q are not strongly bisimilar, as p can per-
form an a-transition into a subsequent state, where it can perform no further
transitions, whereas q can always perform two a-transitions in sequence.

p

p1

p2

p3

q

q1

q2

a a

a

a

a

Strong bisimilarity is the finest extensional behavioural equivalence, because
all actions are thought of as observable. An example of a coarser equivalence
is weak bisimilarity, which treats the aforementioned hidden action τ as
unobservable. However, weak bisimilarity is of no further relevance for this
thesis and the interested reader is referred to [AILS07, Chapter 3.4].
The notion of strong bisimilarity can be formalised through strong bisim-
ulation (SB) relations, introduced originally by David Park in [Par81]. A
binary relation R over the set of processes Proc is an SB iff for all (p, q) ∈ R:

∀p′ ∈ Proc, α ∈ Act . p
α−→ p′ −→ ∃q′ ∈ Proc. q

α−→ q′ ∧ (p′, q′) ∈ R, and

∀q′ ∈ Proc, α ∈ Act . q
α−→ q′ −→ ∃p′ ∈ Proc. p

α−→ p′ ∧ (p′, q′) ∈ R.

Isabelle

Strong bisimulations are straightforward to formalise in Isabelle, using the
‘curried’ definition approach discussed in appendix A.

context lts begin

— strong bisimulation
definition SB :: 〈('s ⇒ 's ⇒ bool) ⇒ bool〉

where 〈SB R ≡ ∀ p q. R p q −→
(∀ p' α. p 7−→α p' −→ (∃ q'. (q 7−→α q') ∧ R p' q')) ∧
(∀ q' α. q 7−→α q' −→ (∃ p'. (p 7−→α p') ∧ R p' q'))〉

Two processes p and q are then strongly bisimilar iff there is an SB that
contains the pair (p, q).
definition strongly_bisimilar :: 〈's ⇒ 's ⇒ bool〉

(〈_ ↔ _〉 [70, 70] 70)
where 〈p ↔ q ≡ ∃ R. SB R ∧ R p q〉

The following corollaries are immediate consequences of these definitions.
corollary strongly_bisimilar_step:

assumes
〈p ↔ q〉

shows
〈p 7−→a p' =⇒ (∃ q'. (q 7−→a q') ∧ p' ↔ q')〉

〈q 7−→a q' =⇒ (∃ p'. (p 7−→a p') ∧ p' ↔ q')〉

〈proof 〉

corollary strongly_bisimilar_symm:
assumes 〈p ↔ q〉

shows 〈q ↔ p〉

〈proof 〉

end — context lts

2.3 Hennessy-Milner Logic

In their seminal paper [HM85], Matthew Hennessy and Robin Milner present
a modal-logical characterisation of strong bisimilarity (although they do not
call it that), by process properties: ‘two processes are equivalent if and only
if they enjoy the same set of properties.’ These properties are expressed as
terms of a modal-logical language, consisting merely of (finite) conjunction,
negation, and a family of modal possibility operators. This language is
known today as Hennessy-Milner logic (HML), with formulas ϕ defined by
the following grammar (where α ranges over the set of actions Act):

ϕ ::= tt | ϕ1 ∧ ϕ2 | ¬ϕ | 〈α〉ϕ

The semantics (on LTS processes) is given as follows: all processes satisfy tt,
ϕ1 ∧ ϕ2 is satisfied if both ϕ1 and ϕ2 are satisfied, ¬ϕ is satisfied if ϕ is
not satisfied, and 〈α〉ϕ is satisfied by a process if it is possible to perform
an α-transition into a process that satisfies ϕ.
[HM85] also contains the proof that this modal-logical characterisation of
strong bisimilarity coincides with a characterisation that is effectively the
same as the one we saw in section 2.2 using strong bisimulations. Although
they use different terminology, their result can be summarised as follows:
for image-finite LTSs, two processes are strongly bisimilar iff they satisfy

the same set of HML formulas. We call this the modal characterisation of
strong bisimilarity.
Let the cardinality of conjunction be the maximally allowed cardinality of
sets of (sub-)formulas conjoined within a formula (for a given variant of
HML). For the simple variant above, conjunction has finite cardinality. By
allowing for conjunction of arbitrary cardinality (infinitary HML), the modal
characterisation of strong bisimilarity can be proved for arbitrary LTSs. This
is done in appendix B.
In this section, however, conjunction is constrained to be of countable car-
dinality, as this turned out to be significantly easier to deal with in the
upcoming proofs. The modal characterisation of strong bisimilarity, then,
works for LTSs that are image-countable, as we shall see below.
Formulas ϕ are given by the following grammar, where I ranges over all
subsets of the natural numbers:

ϕ ::=
∧
i∈I

ϕi | ¬ϕ | 〈α〉ϕ

The semantics of HML formulas on LTSs are as above, with the alteration
that a process satisfies

∧
i∈I ϕi iff it satisfies ϕi for all i ∈ I.

Additional logical constants can be added as ‘syntactic sugar’:

tt ≡
∧

i∈∅ ϕi

ff ≡ ¬tt∨
i∈I

ϕi ≡ ¬
∧
i∈I
¬ϕi

Isabelle

Syntax

By definition of countability, all countable sets of formulas can be given
by {ϕi}i∈I =: Φ for some I ⊆ N (then

∧
i∈I ϕi shall correspond to

∧
Φ).

Therefore, the following data type (parameterised by the type of actions 'a)
formalises the definition of HML formulas above (cset is the type constructor
for countable sets; acset and rcset are the type morphisms between the
types set and cset; more details below).
I abstained from assigning the constructors a more readable symbolic nota-
tion because of the ambiguities and name clashes that would ensue in up-
coming sections. The symbolic notations after the constructors below are
just code comments.

datatype ('a)HML_formula =
HML_conj 〈('a)HML_formula cset〉 —

∧
Φ

| HML_neg 〈('a)HML_formula〉 — ¬ϕ
| HML_poss 〈'a〉 〈('a)HML_formula〉 — 〈α〉ϕ

The following abbreviations introduce useful constants as syntactic sugar
(where cimage HML_neg Φ corresponds to {¬ϕ | ϕ ∈ Φ}).

abbreviation HML_true :: 〈('a)HML_formula〉 — tt
where 〈HML_true ≡ HML_conj (acset ∅)〉

abbreviation HML_false :: 〈('a)HML_formula〉 — ff
where 〈HML_false ≡ HML_neg HML_true〉

abbreviation HML_disj :: 〈('a)HML_formula cset ⇒ ('a)HML_formula〉 —
∨
Φ

where 〈HML_disj Φ ≡ HML_neg (HML_conj (cimage HML_neg Φ))〉

Aside: The Type of Countable Sets

Since sets set and countable sets cset are different types in Isabelle, they
have different membership relation terms. We introduce the following nota-
tion for membership of countable sets.

notation cin (〈_ ∈c _〉 [100, 100] 100)

The following propositions should clarify how the type constructor cset
and its morphisms are used. Note how the first proposition requires the
assumption countable X, whereas the second one does not.

proposition
fixes X::〈'x set〉

assumes 〈countable X〉

shows 〈x ∈ X ⇐⇒ x ∈c acset X〉

〈proof 〉
proposition

fixes X::〈'x cset〉

shows 〈x ∈c X ⇐⇒ x ∈ rcset X〉

〈proof 〉

Semantics

The semantic satisfaction relation is formalised by the following function.
Since the relation is not monotonic (due to negation terms), it cannot be
directly defined in Isabelle as an inductive predicate, so we use the function
command instead. This, then, requires us to prove that the function is well-
defined (i.e. the function definition completely and compatibly covers all
constructors of our data type) and total (i.e. it always terminates). It is
easy to see that the former is the case for the function below.

context lts begin

function HML_sat :: 〈's ⇒ ('a)HML_formula ⇒ bool〉

(〈_ |= _〉 [50, 50] 50)
where
HML_sat_conj: 〈(p |= HML_conj Φ) = (∀ ϕ. ϕ ∈c Φ −→ p |= ϕ)〉

| HML_sat_neg: 〈(p |= HML_neg ϕ) = (¬ p |= ϕ)〉

| HML_sat_poss: 〈(p |= HML_poss α ϕ) = (∃ p'. p 7−→α p' ∧ p' |= ϕ)〉

〈proof 〉

In order to prove that the function always terminates, we need to show
that each sequence of recursive invocations reaches a base case2 after finitely
many steps. We do this by proving that the relation between process-formula
pairs given by the recursive definition of the function is (contained within)
a well-founded relation. A relation R ⊆ X×X is called well-founded if each
non-empty subset X ′ ⊆ X has a minimal element m that is not ‘R-greater’
than any element ofX ′, i.e. ∀x ∈ X ′. (x,m) /∈ R. A property of well-founded
relations is that all descending chains (x0, x1, x2, . . . ) (with (xi, xi+1) ∈ R)
starting at any element x0 ∈ X are finite. This, then, implies that each
sequence of recursive invocations terminates after finitely many steps.
These proofs were inspired by the Isabelle formalisations done in [WEP+16].

inductive_set HML_wf_rel :: 〈('s × ('a)HML_formula) rel〉

where
〈ϕ ∈c Φ =⇒ ((p, ϕ), (p, HML_conj Φ)) ∈ HML_wf_rel〉

| 〈((p, ϕ), (p, HML_neg ϕ)) ∈ HML_wf_rel〉

| 〈((p, ϕ), (p', HML_poss α ϕ)) ∈ HML_wf_rel〉

lemma HML_wf_rel_is_wf: 〈wf HML_wf_rel〉

〈proof 〉

termination HML_sat using HML_wf_rel_is_wf
by (standard, (simp add: HML_wf_rel.intros)+)

The semantic clauses for our additional constants are now easily derivable.

lemma HML_sat_true:
shows 〈(p |= HML_true) = True〉

〈proof 〉
lemma HML_sat_false:

shows 〈(p |= HML_false) = False〉

〈proof 〉
lemma HML_sat_disj:

shows 〈(p |= HML_disj Φ) = (∃ ϕ. ϕ ∈c Φ ∧ p |= ϕ)〉

〈proof 〉

2For our satisfaction function, the recursive base case is, of course, the empty conjunc-
tion, since ∀ϕ. ϕ ∈ ∅ −→ p � ϕ is a tautology.

Modal Characterisation of Strong Bisimilarity

First, we introduce HML-equivalence as follows.

definition HML_equivalent :: 〈's ⇒ 's ⇒ bool〉

where 〈HML_equivalent p q ≡ (∀ ϕ. (p |= ϕ) ←→ (q |= ϕ))〉

Since formulas are closed under negation, the following lemma holds.

lemma distinguishing_formula:
assumes 〈¬ HML_equivalent p q〉

shows 〈∃ ϕ. p |= ϕ ∧ ¬ q |= ϕ〉

〈proof 〉

HML-equivalence is clearly symmetrical.

lemma HML_equivalent_symm:
assumes 〈HML_equivalent p q〉

shows 〈HML_equivalent q p〉

〈proof 〉

We can now formally prove the modal characterisation of strong bisimilarity,
i.e.: two processes are HML-equivalent iff they are strongly bisimilar. The
proof is borrowed from [AILS07, theorem 5.1].
I chose to include these proofs in the thesis document, because they translate
quite beautifully, in my opinion, and are comparatively short.
We show the =⇒-case first, by induction over ϕ.

lemma strong_bisimilarity_implies_HM_equivalence:
assumes 〈p ↔ q〉 〈p |= ϕ〉

shows 〈q |= ϕ〉

using assms
proof (induct ϕ arbitrary: p q)

case (HML_conj Φ)
then show ?case

using HML_sat_conj by (meson cin.rep_eq)
next

case (HML_neg ϕ)
then show ?case

using HML_sat_neg strongly_bisimilar_symm by meson
next

case (HML_poss α ϕ)
then show ?case

using HML_sat_poss strongly_bisimilar_step(1) by meson
qed

Before we can show the⇐=-case, we need to prove the following lemma: for
some binary predicate P , if for every element a of a set A, there exists an
element x such that P (a, x) is true, then we can obtain a set X that contains
these x (for all a ∈ A) and has the same cardinality as A.
Since more than one x might exist for each a such that P (a, x) is true, the
set {x | a ∈ A ∧ P (a, x)} might have greater cardinality than A. In order
to obtain a set X of same cardinality as A, we need to invoke the axiom of
choice in our proof.

lemma obtaining_set:
assumes

〈∀ a ∈ A. ∃ x. P a x〉

〈countable A〉

obtains X where
〈∀ a ∈ A. ∃ x ∈ X. P a x〉

〈∀ x ∈ X. ∃ a ∈ A. P a x〉

〈countable X〉

proof
— the SOME operator (Hilbert’s selection operator ε) invokes the axiom of choice
define xm where 〈xm ≡ λ a. SOME x. P a x〉

define X where 〈X ≡ {xm a | a. a ∈ A}〉

show 〈∀ a∈A. ∃ x∈X. P a x〉

using X_def xm_def assms(1) 〈proof 〉
show 〈∀ x ∈ X. ∃ a∈A. P a x〉

using X_def xm_def assms(1) 〈proof 〉
show 〈countable X〉

using X_def xm_def assms(2) 〈proof 〉
qed

We can now show, assuming image-countability of the given LTS, that HML-
equivalence is a strong bisimulation. The proof utilises classical contradic-
tion: if HML-equivalence were no strong bisimulation, there would be some
processes p and q that are HML-equivalent, with p

α−→ p′ for some p′ (i.e.
p′ ∈ Der(p, α)), but for all q′ ∈ Der(q, α), p′ and q′ are not HML-equivalent.
Then, for each q′ ∈ Der(q, α), there would be a distinguishing formula ϕq′

which p′ satisfies but q′ does not. Using our obtaining_set lemma, we can
obtain the set Φ = {ϕq′}q′∈Der(q,α), which is countable, since Der(q, α) is
countable, by the image-countability assumption. Since we allow for con-
junction of countable cardinality,

∧
Φ is a valid formula. By construction,

p can make an α-transition into a state that satisfies
∧
Φ (i.e. p � 〈α〉

∧
Φ),

whereas q cannot (i.e. q 6� 〈α〉
∧
Φ). This is a contradiction, since, by as-

sumption, p and q are HML-equivalent. Therefore, HML-equivalence must
be a strong bisimulation.

lemma HML_equivalence_is_SB:
assumes

〈image_countable〉

shows
〈SB HML_equivalent〉

proof -
{

fix p q p' α
assume 〈HML_equivalent p q〉

assume 〈p 7−→α p'〉

assume 〈∀ q' ∈ Der(q, α). ¬ HML_equivalent p' q'〉

hence exists_ϕq': 〈∀ q' ∈ Der(q, α). ∃ϕ. p' |= ϕ ∧ ¬ q' |= ϕ〉

using distinguishing_formula by blast

from 〈image_countable〉 have 〈countable Der(q, α)〉

using image_countable_def by simp

from obtaining_set[
where ?A = 〈Der(q, α)〉

and ?P = 〈λ q' ϕ. p' |= ϕ ∧ ¬ q' |= ϕ〉,
OF exists_ϕq' 〈countable Der(q, α)〉]

obtain Φ where *:
〈∀ϕ ∈ Φ. ∃ q' ∈ Der(q, α). p' |= ϕ ∧ ¬ q' |= ϕ〉

〈∀ q' ∈ Der(q, α). ∃ϕ ∈ Φ. p' |= ϕ ∧ ¬ q' |= ϕ〉

〈countable Φ〉

by (this, blast+)

have 〈p |= HML_poss α (HML_conj (acset Φ))〉

using 〈p 7−→α p'〉 *(1,3) HML_sat.simps(1,3)
acset_inverse mem_Collect_eq cin.rep_eq

by metis

moreover have 〈¬ q |= HML_poss α (HML_conj (acset Φ))〉

using *(2,3) cin.rep_eq
by fastforce

ultimately have False
using 〈HML_equivalent p q〉

unfolding HML_equivalent_def
by meson

}

— We showed the case for p 7−→α p', but not q 7−→α q'.
— Clearly, this case is covered by the symmetry of HML-equivalence.
from this show 〈SB HML_equivalent〉 unfolding SB_def

using HML_equivalent_symm by blast
qed

We can now conclude the modal characterisation of strong bisimilarity.

theorem modal_characterisation_of_strong_bisimilarity:
assumes 〈image_countable〉

shows 〈(p ↔ q) ⇐⇒ (∀ ϕ. p |= ϕ ←→ q |= ϕ)〉

proof
show 〈(p ↔ q) =⇒ ∀ϕ. (p |= ϕ) = (q |= ϕ)〉

using strong_bisimilarity_implies_HM_equivalence
strongly_bisimilar_symm

by blast
next

show 〈∀ϕ. (p |= ϕ) = (q |= ϕ) =⇒ (p ↔ q)〉

using HML_equivalence_is_SB[OF assms]
HML_equivalent_def strongly_bisimilar_def

by blast
qed

end — of context lts

2.4 Labelled Transition Systems with Time-Outs

In addition to the hidden action τ , labelled transition systems with time-
outs (LTSt) [vG21] include the time-out action t as another special action;
t-transitions can only be performed when no other (non–time-out) transition
is allowed in a given environment. The rationale is that, in this model, all
transitions that are facilitated by or independent of the environment happen
instantaneously, and only when no such transition is possible, time elapses
and the system is idle. However, since the passage of time is not quantified
here, the system does not have to take a time-out transition in such a case;
instead, the environment can spontaneously change its set of allowed actions
(corresponding to a time-out on the part of the environment). Thus, the
resolution of an idling period is non-deterministic.
In most works on LTSs, the actions which the environment allows in any
given moment are usually not modelled explicitly; an (often implicit) re-
quirement for any property of the system is that it should hold for arbitrary
(and arbitrarily changing) environments. The introduction of time-outs ne-
cessitates an explicit consideration of the environment, as the possibility of
a transition not only depends on whether its labelling action is currently
allowed, but potentially on the set of all actions currently allowed by the
environment. This is why, in previous sections, I have put unusual emphasis
on the actions that are allowed or blocked by the system’s environment in
a given moment. Henceforth, I will refer to ‘environments allowing exactly
the actions in X’ simply as ‘environments X’.

Example The process p can perform an a-transition in environments al-
lowing the action a and a t-transition in environments blocking a. On the
other hand, the process q cannot perform a t-transition in any environment,
since the τ -transition will always be performed immediately.

p

p1 p2

q

q1 q2

a t τ t

Furthermore, since t-transitions (as well as τ -transitions) are hidden, they
cannot trigger a change of the environment, so some states may only ever
be entered in certain environments.

Example The process p can perform a t-transition only in environments
blocking a. Therefore, the subsequent state p2 must be entered in such an
environment. The τ -transition is now the only possible transition and will
always be performed immediately.

p

p1 p2

p3 p4

a t

τ a

On the other hand, transitions with labels other than τ and t require an in-
teraction with the environment, and therefore are detectable and can trigger
a change in the allowed actions of the environment.

Example Only in environments blocking a, p can make a t-transition to
p2. However (if b is allowed), the performance of the b-transition into p3 may
trigger a change of the environment, so it is possible that p3 could perform
its a-transition.

p

p1 p2 p3

p4 p5

a t

b

τ a

These semantics of LTSts induce a novel notion of behavioural equivalence,
which will be discussed in the next section.

Practical Example

As in section 2.1, we shall consider a real-world machine that may be mod-
elled as an LTSt. Continuing with our example, let us imagine that our
simple vending machine ejects the coin if no snack has been selected after a
certain amount of time. We can attempt to model the machine with this ex-
tended behaviour as an LTS, where the coin ejection requires no interaction
and is therefore also modelled as a τ -transition.

p

p1

p2
p3

p4

coin

choc
nuts

crisps

τ
τ

τ

τ

However, this LTS also models a machine that randomly ejects coins right
after insertion. In order to distinguish these behaviours, we need a
t-transition along with LTSt semantics.

p

p1

p′1

p2
p3

p4

coin

choc
nuts

crisps

t

τ
τ

τ

τ

Isabelle

We extend LTSs with hidden actions (lts_tau) by the special action t.
We have to explicitly require (/assume) that τ 6= t; when instantiating the
locale lts_timeout and specifying a concrete type for the type variable 'a,
this assumption must be (proved to be) satisfied.

locale lts_timeout = lts_tau tran τ
for tran :: 's ⇒ 'a ⇒ 's ⇒ bool
(_ 7−→_ _ [70, 70, 70] 80)
and τ :: 'a +

fixes t :: 'a
assumes tau_not_t: 〈τ 6= t〉

begin

We define the set of (relevant) visible actions (denoted by A ⊆ Act) as the
set of all actions that are not hidden and that are labels of some transition
in the given LTS.

definition visible_actions :: 〈'a set〉

where 〈visible_actions
≡ {a. (a 6= τ) ∧ (a 6= t) ∧ (∃ p p'. p 7−→a p')}〉

The formalisations in upcoming sections will often involve the type 'a set
option, which has values of the form None and Some X for some X :: 'a set.
We will use the metavariable XoN (for ‘X or None’). The following abbreviation
will be useful in these situations.

abbreviation some_visible_subset :: 〈'a set option ⇒ bool〉

where 〈some_visible_subset XoN
≡ ∃ X. XoN = Some X ∧ X ⊆ visible_actions〉

The initial actions of a process (I(p) in [vG20]) are the actions for which
the process has a transition it can perform immediately (if facilitated by the
environment), i.e. it is not a t-transition.

definition initial_actions :: 〈's ⇒ 'a set〉

where 〈initial_actions(p)
≡ {α. (α ∈ visible_actions ∨ α = τ) ∧ (∃ p'. p 7−→α p')}〉

In [vG20], the term I(p)∩ (X ∪ {τ}) = ∅ is used a lot, which expresses that
there are no immediate transitions the process p can perform (i.e. it is idle)
in environments X.

abbreviation idle :: 〈's ⇒ 'a set ⇒ bool〉

where 〈idle p X ≡ initial_actions(p) ∩ (X∪{τ}) = ∅〉

The following corollary is an immediate consequence of this definition.
corollary idle_no_derivatives:

assumes
〈idle p X〉

〈X ⊆ visible_actions〉

〈α ∈ X∪{τ}〉

shows
〈@ p'. p 7−→α p'〉

〈proof 〉

end — of locale lts_timeout

2.5 Reactive Bisimilarity

In the examples of the previous section, we saw that there are LTSts with
transitions that can never be performed or that can only be performed in
certain environments. The behavioural equivalence implied hereby is defined
in [vG20] as strong reactive bisimilarity.

Example The processes p and q are behaviourally equivalent for LTSt
semantics, i.e. strongly reactive bisimilar.

p

p1 p2

p3 p4

a t

τ a

q

q1 q2

q3

a t

τ

Strong Reactive Bisimulations

Van Glabbeek introduces several characterisations of this equivalence, be-
ginning with strong reactive bisimulation (SRB) relations. These differ from
strong bisimulations in that the relations contain not only pairs of pro-
cesses, (p, q), but additionally triples consisting of two processes and a set
of actions, (p,X, q). The following definition of SRBs is quoted, with minor
adaptations, from [vG20, definition 1]:
A strong reactive bisimulation is a symmetric relation

R ⊆ (Proc × P(A)× Proc) ∪ (Proc × Proc)

(meaning that (p,X, q)∈R ⇐⇒ (q,X, p)∈R and (p, q)∈R ⇐⇒ (q, p)∈R),
such that,

for all (p, q) ∈ R:

1. if p τ−→ p′, then ∃q′ such that q τ−→ q′ and (p′, q′) ∈ R,

2. (p,X, q) ∈ R for all X ⊆ A,

and for all (p,X, q) ∈ R:

3. if p a−→ p′ with a ∈ X, then ∃q′ such that q a−→ q′ and (p′, q′) ∈ R,

4. if p τ−→ p′, then ∃q′ such that q τ−→ q′ and (p′, X, q′) ∈ R,

5. if I(p) ∩ (X ∪ {τ}) = ∅, then (p, q) ∈ R, and

6. if I(p) ∩ (X ∪ {τ}) = ∅ and p
t−→ p′, then ∃q′ such that q t−→ q′

and (p′, X, q′) ∈ R.

We can derive the following intuitions: an environment can either be stable,
allowing a specific set of actions, or indeterminate. Indeterminate environ-
ments cannot facilitate any transitions, but they can stabilise into arbitrary
stable environments. This is expressed by clause 2. Hence, X-bisimilarity
is behavioural equivalence in stable environments X, and reactive bisimil-
arity is behavioural equivalence in indeterminate environments (and thus in
arbitrary stable environments).
Since only stable environments can facilitate transitions, there are no clauses
involving visible action transitions for (p, q) ∈ R. However, τ -transitions can
be performed regardless of the environment, hence clause 1.
At this point, it is important to discuss what exactly it means for an action
to be visible or hidden in this context: as we saw in the last section, the
environment cannot react (change its set of allowed actions) when the system
performs a τ - or a t-transition, since these are hidden actions. However,
since we are talking about a strong bisimilarity (as opposed to e.g. weak
bisimilarity), the performance of τ - or t-transitions is still relevant when
examining and comparing the behavior of systems.
With that, we can look more closely at the remaining clauses: in clause 3,
given (p,X, q) ∈ R, for p

a−→ p′ with a ∈ X, we require for the ‘mirroring’
state q′ that (p′, q′) ∈ R, because a is a visible action and the transition can
thus trigger a change of the environment;3 on the other hand, in clause 4,
for p

τ−→ p′, and in clause 6, for p
t−→ p′, we require (p′, X, q′) ∈ R, because

these actions are hidden and cannot trigger a change of the environment.
Lastly, clause 5 formalises the possibility of the environment timing out (i.e.
turning into an indeterminate environment) instead of the system.
These intuitions also form the basis for the process mapping which will be
presented in section 3.1.

3This is why van Glabbeek talks about triggered environments rather than indeterm-
inate ones. I will use both terms interchangeably.

Strong Reactive/X-Bisimilarity

Two processes p and q are strongly reactive bisimilar (p↔r q) iff there is an
SRB containing (p, q), and strongly X-bisimilar (p↔X

r q), i.e. equivalent in
environments X, when there is an SRB containing (p,X, q).

Generalised Strong Reactive Bisimulations

Another characterisation of reactive bisimilarity uses generalised strong re-
active bisimulation (GSRB) relations [vG20, definition 3], defined over the
same set as SRBs. It is proved that both characterisations do, in fact,
characterise the same equivalence. More details will be discussed in the
formalisation below.

Isabelle

We first formalise both SRB and GSRB relations (as well as strong reactive
bisimilarity, defined by the existence of an SRB, as above), and then replicate
the proof of their correspondence.

Strong Reactive Bisimulations

SRB relations are defined over the set

(Proc × P(A)× Proc) ∪ (Proc × Proc).

As can be easily seen, this set it isomorphic to

(Proc × (P(A) ∪ {⊥})× Proc),

which is a subset of

(Proc × (P(Act) ∪ {⊥})× Proc).

This last set can now be easily formalised in terms of a type, where we
formalise P(Act) ∪ {⊥} as 'a set option.
The fact that SRBs are defined using the power set of visible actions (A),
whereas our type uses all actions (Act / 'a), is handled by the first line of
the definition below. The second line formalises that symmetry is required
by definition. All other lines are direct formalisations of the clauses of the
original definition.

context lts_timeout begin

— strong reactive bisimulation [vG20, definition 1]
definition SRB :: 〈('s ⇒ 'a set option ⇒ 's ⇒ bool) ⇒ bool〉

where 〈SRB R ≡
(∀ p X q. R p (Some X) q −→ X ⊆ visible_actions) ∧
(∀ p XoN q. R p XoN q −→ R q XoN p) ∧

(∀ p q. R p None q −→
(∀ p'. p 7−→τ p' −→ (∃ q'. (q 7−→τ q') ∧ R p' None q')) ∧
(∀ X ⊆ visible_actions. (R p (Some X) q))) ∧

(∀ p X q. R p (Some X) q −→
(∀ p' a. p 7−→a p' ∧ a ∈ X −→ (∃ q'. (q 7−→a q') ∧

R p' None q')) ∧
(∀ p'. p 7−→τ p' −→ (∃ q'. (q 7−→τ q') ∧ R p' (Some X) q')) ∧
(idle p X −→ R p None q) ∧
(∀ p'. idle p X ∧ (p 7−→t p') −→ (∃ q'. q 7−→t q' ∧

R p' (Some X) q')))〉

Strong Reactive/X-Bisimilarity

Van Glabbeek differentiates between strong reactive bisimilarity ((p, q) ∈ R
for an SRB R) and strong X-bisimilarity ((p,X, q) ∈ R for an SRB R).

definition strongly_reactive_bisimilar :: 〈's ⇒ 's ⇒ bool〉

(〈_ ↔r _〉 [70, 70] 70)
where 〈p ↔r q ≡ ∃ R. SRB R ∧ R p None q〉

definition strongly_X_bisimilar :: 〈's ⇒ 'a set ⇒ 's ⇒ bool〉

(〈_ ↔r
_ _〉 [70, 70, 70] 70)

where 〈p ↔r
X q ≡ ∃ R. SRB R ∧ R p (Some X) q〉

For the upcoming proofs, it is useful to combine both reactive and X-
bisimilarity into a single relation.

definition strongly_reactive_or_X_bisimilar
:: 〈's ⇒ 'a set option ⇒ 's ⇒ bool〉

where 〈strongly_reactive_or_X_bisimilar p XoN q
≡ ∃ R. SRB R ∧ R p XoN q〉

Obviously, then, these relations coincide accordingly.

corollary 〈p ↔r q ⇐⇒ strongly_reactive_or_X_bisimilar p None q〉

〈proof 〉
corollary 〈p ↔r

X q ⇐⇒ strongly_reactive_or_X_bisimilar p (Some X) q〉

〈proof 〉

Generalised Strong Reactive Bisimulations

Since GSRBs are defined over the same set as SRBs, the same considerations
concerning the type and the clauses of the definition as above hold.

— generalised strong reactive bisimulation [vG20, definition 3]
definition GSRB :: 〈('s ⇒ 'a set option ⇒ 's ⇒ bool) ⇒ bool〉

where 〈GSRB R ≡
(∀ p X q. R p (Some X) q −→ X ⊆ visible_actions) ∧
(∀ p XoN q. R p XoN q −→ R q XoN p) ∧

(∀ p q. R p None q −→
(∀ p' a. p 7−→a p' ∧ a ∈ visible_actions ∪ {τ} −→
(∃ q'. q 7−→a q' ∧ R p' None q')) ∧

(∀ X p'. idle p X ∧ X ⊆ visible_actions ∧ p 7−→t p' −→
(∃ q'. q 7−→t q' ∧ R p' (Some X) q'))) ∧

(∀ p Y q. R p (Some Y) q −→
(∀ p' a. a ∈ visible_actions ∧ p 7−→a p' ∧ (a∈Y ∨ idle p Y) −→
(∃ q'. q 7−→a q' ∧ R p' None q')) ∧

(∀ p'. p 7−→τ p' −→
(∃ q'. q 7−→τ q' ∧ R p' (Some Y) q')) ∧

(∀ p' X. idle p (X∪Y) ∧ X ⊆ visible_actions ∧ p 7−→t p' −→
(∃ q'. q 7−→t q' ∧ R p' (Some X) q')))〉

GSRBs characterise strong reactive/X-bisimilarity

[vG20, proposition 4] reads (notation adapted): ‘p ↔r q iff there exists a
GSRB R with (p, q) ∈ R. Likewise, p↔X

r q iff there exists a GSRB R with
(p,X, q) ∈ R.’ We shall now replicate the proof of this proposition. First,
we prove that each SRB is a GSRB (by showing that each SRB satisfies all
clauses of the definition of GSRBs).

lemma SRB_is_GSRB:
assumes 〈SRB R〉

shows 〈GSRB R〉

〈proof 〉

Then, we show that each GSRB can be extended to yield an SRB. First,
we define this extension. Generally, GSRBs can be smaller than SRBs
when proving reactive bisimilarity of processes, because they require triples
(p,X, q) only after encountering t-transitions, whereas SRBs require these
triples for all processes and all environments. Furthermore, some process
pairs (p, q) related to environment time-outs are also omitted in GSRBs.
These tuples are re-added by this extension.

definition GSRB_extension
:: 〈('s⇒'a set option⇒'s ⇒ bool)⇒('s⇒'a set option⇒'s ⇒ bool)〉

where 〈(GSRB_extension R) p XoN q ≡
(R p XoN q)
∨ (some_visible_subset XoN ∧ R p None q)
∨ ((XoN = None ∨ some_visible_subset XoN)
∧ (∃ Y. R p (Some Y) q ∧ idle p Y))〉

Now we show that this extension does, in fact, yield an SRB (again, by
showing that all clauses of the definition of SRBs are satisfied).

lemma GSRB_extension_is_SRB:
assumes

〈GSRB R〉

shows
〈SRB (GSRB_extension R)〉 (is 〈SRB ?R_ext〉)

〈proof 〉

Finally, we can conclude the following:

lemma GSRB_whenever_SRB:
shows 〈(∃ R. GSRB R ∧ R p XoN q) ⇐⇒ (∃ R. SRB R ∧ R p XoN q)〉

〈proof 〉

This, now, directly implies that GSRBs do charactarise strong reactive/X-
bisimilarity.

proposition GSRBs_characterise_strong_reactive_bisimilarity:
〈p ↔r q ⇐⇒ (∃ R. GSRB R ∧ R p None q)〉

using GSRB_whenever_SRB strongly_reactive_bisimilar_def by blast

proposition GSRBs_characterise_strong_X_bisimilarity:
〈p ↔r

X q ⇐⇒ (∃ R. GSRB R ∧ R p (Some X) q)〉

using GSRB_whenever_SRB strongly_X_bisimilar_def by blast

end — of context lts_timeout

As a little meta-comment, I would like to point out that van Glabbeek’s
proof spans a total of five lines (‘Clearly, […]’, ‘It is straightforward to check
[…]’), whereas the Isabelle proof takes up around 250 lines of code. This just
goes to show that for things which are clear and straightforward for humans,
it might require quite some effort to ‘explain’ them to a computer.

2.6 Hennessy-Milner Logic with Time-Outs

In [vG20, section 3], van Glabbeek extends Hennessy-Milner logic by a family
of new modal operators 〈X〉ϕ for X ⊆ A, as well as additional satisfaction
relations �X for X ⊆ A. Intuitively, p � 〈X〉ϕ means that p is idle when
placed in an environment X and p can perform a t-transition into a state
that satisfies ϕ; p �X ϕ means that p satisfies ϕ in environments X.
I will refer to this extension as Hennessy-Milner Logic with Time-Outs
(HMLt) and to 〈X〉 for X ⊆ A as the time-out–possibility operators (to
be distinguished from the ordinary possibility operators 〈α〉 for α ∈ Act).
The precise semantics are given by the following inductive definition of the
satisfaction relation [vG20, section 3] (notation adapted):
p �

∧
i∈I ϕi if ∀i ∈ I. p � ϕi

p � ¬ϕ if p 6� ϕ

p � 〈α〉ϕ with α ∈ A ∪ {τ} if ∃p′. p α−→ p′ ∧ p′ � ϕ

p � 〈X〉ϕ with X ⊆ A if I(p) ∩ (X ∪ {τ}) = ∅ ∧ ∃p′. p t−→ p′ ∧ p′ �X ϕ

p �X
∧

i∈I ϕi if ∀i ∈ I. p �X ϕi

p �X ¬ϕ if p 6�X ϕ

p �X 〈a〉ϕ with a ∈ A if a ∈ X ∧ ∃p′. p a−→ p′ ∧ p′ � ϕ

p �X 〈τ〉ϕ if ∃p′. p τ−→ p′ ∧ p′ �X ϕ

p �X ϕ if I(p) ∩ (X ∪ {τ}) = ∅ ∧ p � ϕ

The same intuitions regarding triggered and stable environments as for the
definition of strong reactive bisimulations in section 2.2 hold. � expresses
that a property holds in indeterminate environments and �X that a property
holds in stable environments X. The last clause expresses the possibility of
stable environments timing out into triggered environments.
Van Glabbeek then also proves that HMLt characterises strong reactive/X-
bisimilarity, i.e. that p↔r q ⇐⇒ (∀ϕ. p � ϕ←→ q � ϕ) and p↔X

r q ⇐⇒
(∀ϕ. p �X ϕ←→ q �X ϕ), where ϕ are formulas of HMLt. A replication of
the proof of this characterisation, however, is not part of this thesis.

Isabelle

The following formalisation is analogous to the one in section 2.3.

datatype ('a)HMLt_formula =
HMLt_conj 〈('a)HMLt_formula cset〉 —

∧
Φ

| HMLt_neg 〈('a)HMLt_formula〉 — ¬ϕ
| HMLt_poss 〈'a〉 〈('a)HMLt_formula〉 — 〈α〉ϕ
| HMLt_time 〈'a set〉 〈('a)HMLt_formula〉 — 〈X〉ϕ

In order to formalise the semantics, I combined both the usual satisfaction
relation � and the environment satisfaction relations �X into one predicate,
which is formalised by the function HMLt_sat below, where p ||=?[None] ϕ

corresponds to p � ϕ and p ||=?[Some X] ϕ corresponds to p �X ϕ.
Note that, in Isabelle code, I use the symbol ||= for all satisfaction relations
in the context of HMLt, whereas I use |= for satisfaction relations in the
context of ordinary HML. This notational nuance will be important when
we examine the relationship between the satisfaction relations of HMLt and
HML in the context of the reduction in section 3.4.
The first four clauses of my formalisation are clearly direct translations of
the clauses for the satisfaction relation � above. It is less easy to see that
the next four clauses do, in fact, correspond to the five clauses for �X .
First, each of the four clauses below includes the requirement that X is a sub-
set of the visible actions; in the original definition, the satisfaction relations
�X are only defined for those X to begin with.
Next, the clause for p ||=?[Some X] (HMLt_poss α ϕ) combines the original
clauses for p �X 〈a〉ϕ and p �X 〈τ〉ϕ.
Lastly and most importantly, the last clause of the original definition, stating
that p �X ϕ if p is idle in environments X and p � ϕ, is added disjunctively
to the cases p ||=?[Some X] (HMLt_poss α ϕ) and p ||=?[Some X] (HMLt_time
Y ϕ); the latter case is not part of the original definition and can only be
true by virtue of the last clause of the original definition, wherefore this is
the only way for this case in the function definition below to be true.
I will show below that this is sufficient to assure that my satisfaction function
satisfies the last clause of the original definition, i.e. that it is not required
to be added disjunctively to the cases p ||=?[Some X] (HMLt_conj Φ) and p
||=?[Some X] (HMLt_neg ϕ).
context lts_timeout begin

function HMLt_sat :: 〈's⇒'a set option⇒('a)HMLt_formula ⇒ bool〉

(〈_ ||=?[_] _〉 [50, 50, 50] 50)
where

〈(p ||=?[None] (HMLt_conj Φ)) =
(∀ ϕ. ϕ ∈c Φ −→ p ||=?[None] ϕ)〉

| 〈(p ||=?[None] (HMLt_neg ϕ)) =
(¬ p ||=?[None] ϕ)〉

| 〈(p ||=?[None] (HMLt_poss α ϕ)) =
((α ∈ visible_actions ∪ {τ}) ∧
(∃ p'. p 7−→α p' ∧ p' ||=?[None] ϕ))〉

| 〈(p ||=?[None] (HMLt_time X ϕ)) =
((X ⊆ visible_actions) ∧ (idle p X) ∧
(∃ p'. p 7−→t p' ∧ p' ||=?[Some X] ϕ))〉

| 〈(p ||=?[Some X] (HMLt_conj Φ)) = (X ⊆ visible_actions ∧
(∀ ϕ. ϕ ∈c Φ −→ p ||=?[Some X] ϕ))〉

| 〈(p ||=?[Some X] (HMLt_neg ϕ)) = (X ⊆ visible_actions ∧
(¬ p ||=?[Some X] ϕ))〉

| 〈(p ||=?[Some X] (HMLt_poss α ϕ)) = (X ⊆ visible_actions ∧
(((α ∈ X) ∧ (∃ p'. p 7−→α p' ∧ p' ||=?[None] ϕ)) ∨
((α = τ) ∧ (∃ p'. p 7−→τ p' ∧ p' ||=?[Some X] ϕ)) ∨
((idle p X) ∧ (p ||=?[None] (HMLt_poss α ϕ)))))〉

| 〈(p ||=?[Some X] (HMLt_time Y ϕ)) = (X ⊆ visible_actions ∧
((idle p X) ∧ (p ||=?[None] (HMLt_time Y ϕ))))〉

〈proof 〉

The well-founded relation used for the termination proof of the satisfaction
function is considerably more difficult due to the last line of the definition
containing the same formula on both sides of the implication (as opposed to
the other lines of the definition, where the premises only contain subformu-
las of the formula in the conclusion). We define two relations R and S, prove
their well-foundedness separately, and show that R O S ⊆ R (where O is re-
lation composition), yielding that the union of R and S is well-founded using
the theorem [[wf R; wf S; R O S ⊆ R]] =⇒ wf (R ∪ S). Further details are
omitted from the thesis document.
termination HMLt_sat 〈proof 〉

We can now introduce the more readable notation (more closely correspond-
ing to the notation in [vG20]) through abbreviations.
abbreviation HMLt_sat_triggered :: 〈's⇒('a)HMLt_formula ⇒ bool〉

(_ ||= _ [50, 50] 50)
where 〈p ||= ϕ ≡ p ||=?[None] ϕ〉

abbreviation HMLt_sat_stable :: 〈's⇒'a set⇒('a)HMLt_formula ⇒ bool〉

(_ ||=[_] _ [70, 70, 70] 80)
where 〈p ||=[X] ϕ ≡ p ||=?[Some X] ϕ〉

Lastly, we show (by induction over ϕ) that the function HMLt_sat does indeed
satisfy the last clause of the original definition.
proposition

assumes
〈X ⊆ visible_actions〉

〈idle p X〉

〈p ||= ϕ〉

shows
〈p ||=[X] ϕ〉

〈proof 〉

As the last clause of van Glabbeek’s definition is the main disparity to the
function definition of HMLt_sat, this proposition gives confidence that the
function does indeed formalise the original definition.
end — of context lts_timeout

The Reductions

In [vG20], various characterisations of reactive bisimilarity are presented,
three of which we have seen in previous sections (SRBs, GSRBs, and the
modal characterisation). Another one introduces environment operators θX ,
which ‘place a process in [a stable] environment that allows exactly the
actions in X to occur’ [vG20, section 4]. The precise semantics are given
by structural operational rules, e.g.: p τ−→ p′ =⇒ θX(p)

τ−→ θX(p′). For the
characterisation of reactive bisimilarity, the definition of another kind of
relations, namely time-out bisimulations, is required.
This inspired me to come up with a mapping (from LTSts to LTSs) that
explicitly models the entire behaviour of the environment and its interaction
with the reactive system. Concretely, the resulting LTS will contain a state
for each state of the original LTSt in every possible environment (including
indeterminate ones). Therefore, the resulting LTS will not be a model of a
reactive system, but of the closed system consisting of the original underlying
system and its environment.
This approach basically treats the environment as an unknown process
placed in parallel with the system; this has also been suggested by van Glab-
beek in [vG20, section 2]. There, however, the action t must still be treated
as a special action with special semantics. For the reduction presented in the
next sections, the entire semantics of LTSts will be incorporated in the map-
ping, where all actions are then treated equally, and so that two processes
of an LTSt are strongly reactive bisimilar iff their corresponding processes
in the mapped LTS are strongly bisimilar. The mapping will be presented
in section 3.1 and the reduction established in section 3.2.
As a natural consequence, a reduction for the satisfaction of HMLt formulas
can be given as well: in section 3.3, I will present a mapping from HMLt
formulas to HML formulas such that, as we will see in section 3.4, a mapped
formula holds in a process of a mapped LTS iff the original formula holds in
the corresponding process of the original LTSt.

3.1 A Mapping for Transition Systems

Let T = (Proc,Act ,→) be an LTSt. Let A = Act \{τ, t}.
In reference to van Glabbeek’s θX -operators, I introduce a family of oper-
ators ϑX with similar but not identical semantics. Additionally, I introduce
the operator ϑ that places a process in an indeterminate environment.
Furthermore, I introduce a family of special actions εX for X ⊆ A that
represent a triggered environment stabilising into an environment X, as
well as a special action tε that represents a time-out of the environment.
We assume that tε /∈ Act and ∀X ⊆ A. εX /∈ Act .
Then we define Tϑ = (Procϑ,Actϑ,→ϑ) with

Procϑ = {ϑ(p) | p ∈ Proc} ∪ {ϑX(p) | p ∈ Proc ∧X ⊆ A},
Actϑ = Act ∪ {tε} ∪ {εX | X ⊆ A},

and →ϑ defined by the following rules:

(1)
p

τ−→ p′

ϑ(p)
τ−→ϑ ϑ(p′)

(2)
ϑ(p)

εX−−→ϑ ϑX(p)
X ⊆ A

(3)
p

a−→ p′

ϑX(p)
a−→ϑ ϑ(p′)

a ∈ X (4)
p

τ−→ p′

ϑX(p)
τ−→ϑ ϑX(p′)

(5)
p 6 α−→ for all α ∈ X ∪ {τ}

ϑX(p)
tε−→ϑ ϑ(p)

(6)
p 6 α−→ for all α ∈ X ∪ {τ} p

t−→ p′

ϑX(p)
t−→ϑ ϑX(p′)

These rules mirror the clauses of the definition of SRBs (cf. section 2.5):
1. τ -transitions can be performed regardless of the environment,
2. indeterminate environments can stabilise into arbitrary stable

environments X for X ⊆ A,
3. facilitated visible transitions can be performed and can trigger a change

in the environment,
4. τ -transitions cannot be observed by the environment and hence cannot

trigger a change,
5. if the underlying system is idle, the environment may time-out and

turn into an indeterminate/triggered environment,
6. if the underlying system is idle and has a t-transition, the transition

may be performed and is not observable by the environment.

Example The LTSt on the left (with Act = {a, τ, t}) gets mapped to the
LTS on the right. States that have no incoming or outgoing transitions other
than εX or tε are omitted. Note how ϑ(p4) is not reachable from ϑ(p).

p

p1 p2

p3 p4

a t

τ a

ϑ(p)

ϑ{a}(p) ϑ∅(p) ϑ(p2)

ϑ(p1) ϑ∅(p2)

ϑ∅(p3)

ϑ{a}(p2) ϑ(p4)

ϑ{a}(p3)

ϑ(p3)

ε{a}
ε∅

tε

a t

τ

ε{a}ε∅

a

τ

tε
ε∅

tε
ε{a}

τε... tε ε... tε

Isabelle

Formalising Procϑ and Actϑ

We specify another locale based on lts_timeout, where the aforementioned
special actions and operators are considered; we call it lts_timeout_mappable.
Since Proc ∩Procϑ = ∅, we introduce a new type variable 'ss for Procϑ; we
use 'a for both Act and Actϑ. We formalise the family of special actions εX
as a mapping ε[_] :: 'a set ⇒ 'a, and the environment operators ϑ/ϑX

as a single mapping ϑ?[_](_) :: 'a set option ⇒ 's ⇒ 'ss.
As for lts_timeout in section 2.4, we require that all special actions are dis-
tinct, formalised by the first set of assumptions distinctness_special_actions.
As an operator, the term ϑX(p) simply refers to the state p in an environ-
ment X; when understood as a mapping, we have to be more careful, since
ϑ?[Some X](p) is now itself a state. Specifically, we have to assume that
ϑ?[_](_) is injective (when restricted to domains where
X ⊆ visible_actions, because ϑX is only defined for those X ⊆ A). Other-
wise, we might have ϑ?[None](p) = ϑ?[None](q) for p 6= q, which is prob-
lematic if e.g. p has a τ -transition, but q does not. The restricted injectivity
of ϑ?[_](_) is formalised as the set of assumptions injectivity_theta.
The same is required for the mapping ε[_], as formalised in the last clause
of the set of assumptions distinctness_special_actions (the restricted in-
jectivity of ε[_] is part of the requirement that all special actions must be
distinct). Again, we only require injectivity for the mapping restricted to
the domain visible_actions. If we required that ε[_] :: 'a set ⇒ 'a were

injective over its entire domain 'a set, we would run into problems, since
such a function cannot exist by Cantor’s theorem.
That such mappings exist is intuitively clear, whence there were no ambigu-
ities when defining them as operators in the prosaic/mathematical section
above. Formalising these mappings in HOL, however, is not so straight-
forward: as operators, we assume that they are only defined for certain
parameters; in HOL, every mapping must be total. For now, we simply
assume that such total functions that formalise these operators exist. Do-
ing this significantly improves the readability of following sections, since
we must only consider the relevant properties of the mappings given by
the assumptions. In appendix C, I give examples for these mappings and
show that, together with these, any lts_timeout can be interpreted as an
lts_timeout_mappable, i.e. every LTSt T can be mapped to an LTS Tϑ.
Lastly, we formalise our requirements tε /∈ Act and ∀X ⊆ A. εX /∈ Act as the
last set of assumptions no_epsilon_in_tran. Technically, these assumptions
only state that the ε-actions do not label any transition of T. However, we
can assume that Act = {α | ∃p, p′. p α−→ p′}, since actions that do not label
any transitions are not relevant to the behaviour of an LTS.

locale lts_timeout_mappable = lts_timeout tran τ t
for tran :: 's ⇒ 'a ⇒ 's ⇒ bool

(_ 7−→_ _ [70, 70, 70] 80)
and τ :: 'a
and t :: 'a +

fixes t_ε :: 'a
and stabilise :: 〈'a set ⇒ 'a〉

(〈ε[_]〉)
and in_env :: 〈'a set option ⇒ 's ⇒ 'ss〉

(〈ϑ?[_]'(_')〉)
assumes
distinctness_special_actions:
〈τ 6= t〉 〈τ 6= t_ε〉 〈t 6= t_ε〉

〈ε[X] 6= τ 〉 〈ε[X] 6= t〉 〈ε[X] 6= t_ε〉

〈X ⊆ visible_actions =⇒ ε[X] = ε[Y] =⇒ X = Y〉

and

injectivity_theta:
〈ϑ?[None](p) 6= ϑ?[Some X](q)〉

〈(ϑ?[None](p) = ϑ?[None](q)) −→ p = q〉

〈X ⊆ visible_actions =⇒
(ϑ?[Some X](p) = ϑ?[Some Y](q)) −→ X = Y ∧ p = q〉

and

no_epsilon_in_tran:
〈¬ p 7−→ε[X] q〉

〈¬ p 7−→t_ε q〉

begin

We can now define abbreviations with notations that correspond more closely
to our operators defined above.
abbreviation triggered_env :: 〈's ⇒ 'ss〉

(〈ϑ'(_')〉)
where 〈ϑ(p) ≡ ϑ?[None](p)〉

abbreviation stable_env :: 〈'a set ⇒ 's ⇒ 'ss〉

(〈ϑ[_]'(_')〉)
where 〈ϑ[X](p) ≡ ϑ?[Some X](p)〉

Formalising →ϑ

We formalise the transition relation of our mapping, given above by the
structural operational rules, as a function tran_theta.1

We use the inductive command, because it allows us to define separate
clauses (as opposed to the definition command). Technically speaking,
however, this inductive definition only has base cases, since none of the
premises involves 7−→ϑ.
It should be easy to see that the clauses below correspond directly to the
rules above. Like in previous sections, we have to take extra care to handle
the requirement X ⊆ visible_actions.
inductive tran_theta :: 〈'ss ⇒ 'a ⇒ 'ss ⇒ bool〉

(〈_ 7−→ϑ_ _〉 [70, 70, 70] 70)
where
triggered_tau:

〈p 7−→τ q =⇒ ϑ(p) 7−→ϑτ ϑ(q)〉

| env_stabilise: 〈X ⊆ visible_actions =⇒
ϑ(p) 7−→ϑε[X] ϑ[X](p)〉

| tran_visible: 〈X ⊆ visible_actions =⇒
a ∈ X =⇒ p 7−→a q =⇒ ϑ[X](p) 7−→ϑa ϑ(q)〉

| stable_tau: 〈X ⊆ visible_actions =⇒
p 7−→τ q =⇒ ϑ[X](p) 7−→ϑτ ϑ[X](q)〉

| env_timeout: 〈X ⊆ visible_actions =⇒
idle p X =⇒ ϑ[X](p) 7−→ϑt_ε ϑ(p)〉

| sys_timeout: 〈X ⊆ visible_actions =⇒
idle p X =⇒ p 7−→t q =⇒ ϑ[X](p) 7−→ϑt ϑ[X](q)〉

Note on Metavariable usage

If not referenced directly by ϑ(p) or ϑX(p), arbitrary states of a mapped LTS
range over P,Q, P ′, Q′, . . . , where P and P ′ are used for states connected
by some transition (i.e. P α−→ϑ P ′), whereas P and Q are used for states
possibly related by some equivalence (e.g. P ↔r Q).

1We use the notation _ 7−→ϑ_ _ instead of the more obvious _ 7−→ϑ_ _ simply
because of better readability.

Generation Lemmas

Lastly, we derive a set of generation lemmas, i.e. lemmas that allow us to
reason backwards: if we know P 7−→ϑα P' and some other information about
P and/or α, we can deduce some information about the other variables as
well as the transitions of the original LTSt.

lemma generation_triggered_transitions:
assumes 〈ϑ(p) 7−→ϑα P'〉

shows 〈(∃ X. α = ε[X] ∧ P' = ϑ[X](p) ∧ X ⊆ visible_actions)
∨ (α = τ ∧ (∃ p'. p 7−→τ p'))〉

〈proof 〉

lemma generation_stable_transitions:
assumes 〈ϑ[X](p) 7−→ϑα P'〉

shows 〈α = t_ε ∨ (∃ p'. p 7−→α p' ∧ (α ∈ X ∨ α = τ ∨ α = t))〉

〈proof 〉

lemma generation_triggered_tau:
assumes 〈ϑ(p) 7−→ϑτ P'〉

shows 〈∃ p'. P' = ϑ(p') ∧ p 7−→τ p'〉

〈proof 〉

lemma generation_env_stabilise:
assumes 〈P 7−→ϑε[X] P'〉

shows 〈∃ p. P = ϑ(p) ∧ P' = ϑ[X](p)〉

〈proof 〉

lemma generation_tran_visible:
assumes 〈ϑ[X](p) 7−→ϑa P'〉 〈a ∈ visible_actions〉

shows 〈a ∈ X ∧ (∃ p'. P' = ϑ(p') ∧ p 7−→a p')〉

〈proof 〉

lemma generation_stable_tau:
assumes 〈ϑ[X](p) 7−→ϑτ P'〉

shows 〈∃ p'. P' = ϑ[X](p') ∧ p 7−→τ p'〉

〈proof 〉

lemma generation_env_timeout:
assumes 〈ϑ[X](p) 7−→ϑt_ε P'〉

shows 〈P' = ϑ(p) ∧ idle p X〉

〈proof 〉

lemma generation_sys_timeout:
assumes 〈ϑ[X](p) 7−→ϑt P'〉

shows 〈∃ p'. P' = ϑ[X](p') ∧ idle p X ∧ p 7−→t p'〉

〈proof 〉

end — of locale lts_timeout_mappable

3.2 Reduction of Bisimilarity

The result of this section will be that two processes p and q of an LTSt T
are strongly reactive bisimilar (strongly X-bisimilar) iff the corresponding
processes ϑ(p) and ϑ(q) (ϑX(p) and ϑX(q)) of Tϑ are strongly bisimilar.
We show the =⇒-direction first. For an SRB R, let

S = {(ϑ(p), ϑ(q)) | (p, q) ∈ R} ∪ {(ϑX(p), ϑX(q)) | (p,X, q) ∈ R}.

We can prove that S is an SB, by showing that the mapping satisfies the
clauses of the definition of SBs, using the fact that R is an SRB as well as
the rules and generation lemmas for →ϑ. Hence, the existence of an SRB
R with (p, q) ∈ R implies the existence of an SB S with (ϑ(p), ϑ(q)) ∈ S
(and similarly for ϑX), so strong reactive/X-bisimilarity in T implies strong
bisimilarity in Tϑ.
Next, we show the ⇐=-direction. Let

R = {(p, q) | ϑ(p)↔ ϑ(q)} ∪ {(p,X, q) | ϑX(p)↔ ϑX(q)}.

We can prove that R is an SRB, again, by showing that all clauses of the
definition are satisfied. Hence, strong bisimilarity of ϑ(p) and ϑ(q) implies
the existence of an SRB R with (p, q) ∈ R (and similarly for ϑX), so strong
bisimilarity in Tϑ implies strong reactive/X-bisimilarity in T.
Thus, we have that strong reactive/X-bisimilarity in T corresponds to strong
bisimilarity in Tϑ.

Isabelle

We begin by interpreting our transition mapping tran_theta as an lts and
call it lts_theta. Therefore, we are handling two separate LTSs: the LTSt T
given by the local context lts_timeout_mappable, and the LTS Tϑ given by
the interpretation lts_theta. When referring to definitions involving the
transition relation of lts_theta, we have to prefix them, e.g. lts_theta.SB
for the definition of strong bisimulations using 7−→ϑ instead of 7−→.
By default, interpretations do not import special notation, so we reassign
strong bisimilarity notation ↔ to lts_theta, since we do not care about
strong bisimilarity in T.
context lts_timeout_mappable begin

interpretation lts_theta: lts tran_theta 〈proof 〉
no_notation local.strongly_bisimilar (〈_ ↔ _〉 [70, 70] 70)
notation lts_theta.strongly_bisimilar (〈_ ↔ _〉 [70, 70] 70)

We can now formalise the proof as described above.

If … (=⇒)

definition SRB_mapping — S
:: 〈('s⇒'a set option⇒'s ⇒ bool) ⇒ ('ss⇒'ss ⇒ bool)〉

where 〈SRB_mapping R P Q ≡
(∃ p q. P = ϑ(p) ∧ Q = ϑ(q) ∧ R p None q) ∨
(∃ p q X. P = ϑ[X](p) ∧ Q = ϑ[X](q) ∧ R p (Some X) q)〉

lemma SRB_mapping_is_SB:
assumes 〈SRB R〉

shows 〈lts_theta.SB (SRB_mapping R)〉 (is 〈lts_theta.SB ?S〉)
〈proof 〉

lemma srby_implies_sby:
assumes 〈p ↔r q〉

shows 〈ϑ(p) ↔ ϑ(q)〉

〈proof 〉
lemma sxby_implies_sby:

assumes 〈p ↔r
X q〉

shows 〈ϑ[X](p) ↔ ϑ[X](q)〉

〈proof 〉

… and only if (⇐=)

definition strong_bisimilarity_mapping — R
:: 〈's⇒'a set option⇒'s ⇒ bool〉

where 〈(strong_bisimilarity_mapping) p XoN q
≡ (XoN = None ∧ (ϑ(p)) ↔ (ϑ(q))) ∨
(∃ X. XoN = Some X ∧ X ⊆ visible_actions ∧
ϑ[X](p) ↔ ϑ[X](q))〉

lemma strong_bisimilarity_mapping_is_SRB:
shows 〈SRB strong_bisimilarity_mapping〉 (is 〈SRB ?R〉)
〈proof 〉

lemma sby_implies_srby:
assumes 〈ϑ(p) ↔ ϑ(q)〉

shows 〈p ↔r q〉

〈proof 〉
lemma sby_implies_sxby:

assumes 〈ϑ[X](p) ↔ ϑ[X](q)〉 〈X ⊆ visible_actions〉

shows 〈p ↔r
X q〉

〈proof 〉

We need to include the assumption X ⊆ visible_actions, since for
¬ X ⊆ visible_actions, ϑ[X](p) and ϑ[X](q) might be identical (since we
do not require injectivity for that subset of the domain), so ϑ[X](p) ↔
ϑ[X](q) would be true, whereas p ↔r

X q would be false (since
X ⊆ visible_actions is part of the definition of SRBs).

Iff (⇐⇒)

theorem strongly_reactive_bisim_iff_triggered_strongly_bisim:
shows 〈p ↔r q ⇐⇒ ϑ(p) ↔ ϑ(q)〉

using sby_implies_srby srby_implies_sby by fast

theorem strongly_X_bisim_iff_stable_strongly_bisim:
assumes 〈X ⊆ visible_actions〉

shows 〈p ↔r
X q ⇐⇒ ϑ[X](p) ↔ ϑ[X](q)〉

using sxby_implies_sby sby_implies_sxby assms by fast

end — of context lts_timeout_mappable

3.3 A Mapping for Formulas

We will now introduce a mapping σ(·) that maps formulas of HMLt to
formulas of HML, in the context of the process mapping from section 3.1,
such that ϑ(p) satisfies σ(ϕ) iff p satisfies ϕ.
Again, we have T = (Proc,Act ,→) and Tϑ = (Procϑ,Actϑ,→ϑ) as defined
in section 3.1, with A = Act \{τ, t}, and we assume that tε /∈ Act and
∀X ⊆ A. εX /∈ Act .
Let σ : (HMLt formulas) −→ (HML formulas) be recursively defined by

σ(
∧

i∈I ϕi) =
∧

i∈I σ(ϕi)

σ(¬ϕ) = ¬σ(ϕ)
σ(〈τ〉ϕ) = 〈τ〉σ(ϕ)
σ(〈α〉ϕ) = 〈α〉σ(ϕ) ∨

〈εA〉〈α〉σ(ϕ) ∨
〈tε〉〈εA〉〈α〉σ(ϕ) if α ∈ A

σ(〈α〉ϕ) = ff if α /∈ A ∪ {τ}
σ(〈X〉ϕ) = 〈εX〉〈t〉σ(ϕ) ∨

〈tε〉〈εX〉〈t〉σ(ϕ) if X ⊆ A

σ(〈X〉ϕ) = ff if X 6⊆ A

This mapping simply expresses the time-out semantics given by the satisfac-
tion relations of HMLt (section 2.6) in terms of ordinary HML evaluated on
our mapped LTS Tϑ. The disjunctive clauses compensate for the additional
environment transitions (ε-actions) that are not present in T.

Isabelle

The implementation of the mapping in Isabelle is rather straightforward,
although some details might not be obvious:
cimage (λ ϕ. σ(ϕ)) Φ is the image of the countable set Φ under the func-
tion λ ϕ. σ(ϕ), so it corresponds to {σ(ϕ) | ϕ ∈ Φ} for countable Φ.
α 6= τ ∧ α 6= t ∧ α 6= t_ε ∧ (∀ X. α 6= ε[X]) corresponds to α ∈ A with
our assumption about there being no ε-actions in Act . Similarly,
α = t ∨ α = t_ε ∨ α = ε[X] corresponds to α /∈ A ∪ {τ}.

context lts_timeout_mappable begin

function HMt_mapping :: 〈('a)HMLt_formula ⇒ ('a)HML_formula〉

(〈σ'(_')〉)
where

〈σ(HMLt_conj Φ) = HML_conj (cimage (λ ϕ. σ(ϕ)) Φ)〉

| 〈σ(HMLt_neg ϕ) = HML_neg σ(ϕ)〉

| 〈α = τ =⇒
σ(HMLt_poss α ϕ) = HML_poss α σ(ϕ)〉

| 〈α 6= τ ∧ α 6= t ∧ α 6= t_ε ∧ (∀ X. α 6= ε[X]) =⇒
σ(HMLt_poss α ϕ) = HML_disj (acset {
HML_poss α σ(ϕ),
HML_poss ε[visible_actions] (HML_poss α σ(ϕ)),
HML_poss t_ε (HML_poss ε[visible_actions] (HML_poss α σ(ϕ)))

})〉

| 〈α = t ∨ α = t_ε ∨ α = ε[X] =⇒
σ(HMLt_poss α ϕ) = HML_false〉

| 〈X ⊆ visible_actions =⇒
σ(HMLt_time X ϕ) = HML_disj (acset {
HML_poss ε[X] (HML_poss t σ(ϕ)),
HML_poss t_ε (HML_poss ε[X] (HML_poss t σ(ϕ)))

})〉

| 〈¬ X ⊆ visible_actions =⇒
σ(HMLt_time X ϕ) = HML_false〉

〈proof 〉

Again, we show that the function terminates using a well-founded relation.

inductive_set sigma_wf_rel :: 〈(('a)HMLt_formula) rel〉

where
〈ϕ ∈c Φ =⇒ (ϕ, HMLt_conj Φ) ∈ sigma_wf_rel〉

| 〈(ϕ, HMLt_neg ϕ) ∈ sigma_wf_rel〉

| 〈(ϕ, HMLt_poss α ϕ) ∈ sigma_wf_rel〉

| 〈(ϕ, HMLt_time X ϕ) ∈ sigma_wf_rel〉

termination HMt_mapping 〈proof 〉

end — of context lts_timeout_mappable

3.4 Reduction of Formula Satisfaction

We will show that, for a process p of an LTSt and a formula ϕ of HMLt,
we have p � ϕ ⇐⇒ ϑ(p) � σ(ϕ) and p �X ϕ ⇐⇒ ϑX(p) � σ(ϕ). The
proof is rather straightforward: we use induction over HMLt formulas and
show that, for each case, the semantics given by van Glabbeek’s satisfaction
relations and those given by the mappings σ and ϑ/ϑX coincide. Due to the
relative complexity of the mapping and the satisfaction relations, the proof
is quite tedious, however.

Isabelle

Similarly to the formalisations in section 3.2, we begin by interpreting our
transition mapping tran_theta as an lts and reassigning notation appropri-
ately (we only care about HML formula satisfaction for Tϑ, not T).
context lts_timeout_mappable begin

interpretation lts_theta: lts tran_theta 〈proof 〉
no_notation local.HML_sat (_ |= _ [50, 50] 50)
notation lts_theta.HML_sat (_ |= _ [50, 50] 50)

We show p ||=?[XoN] ϕ ⇐⇒ ϑ?[XoN](p) |= σ(ϕ) by induction over ϕ. By
using those terms for formula satisfaction and process mappings that handle
both triggered and stable environments, we can handle both situations sim-
ultaneously, which is required due to the interdependence of � and �X . How-
ever, this requires us to consider four cases (each combination of ||=?[XoN1]
and ϑ?[XoN2] for XoN1, XoN2 ∈ {None, Some X}2) per inductive case for ϕ.
Together with the many disjunctive clauses in the mapping, a large number
of cases needs to be considered, leading to a proof spanning roughly 350
lines of Isabelle code.
lemma HMLt_sat_iff_HML_sat:

assumes 〈XoN = None ∨ (XoN = (Some X) ∧ X ⊆ visible_actions)〉

shows 〈p ||=?[XoN] ϕ ⇐⇒ ϑ?[XoN](p) |= σ(ϕ)〉

〈proof 〉

Theorems using nicer notation are immediate consequences of this lemma.
theorem HMLt_sat_triggered_iff_triggered_env_HML_sat:

shows 〈p ||= ϕ ⇐⇒ ϑ(p) |= σ(ϕ)〉

using HMLt_sat_iff_HML_sat by blast
theorem HMLt_sat_stable_iff_stable_env_HML_sat:

assumes 〈X ⊆ visible_actions〉

shows 〈p ||=[X] ϕ ⇐⇒ ϑ[X](p) |= σ(ϕ)〉

using HMLt_sat_iff_HML_sat assms by blast

end — of context lts_timeout_mappable
2Once again, we only consider cases where X ⊆ visible_actions.

Discussion

We have shown that checking strong reactive/X-bisimilarity (in an LTSt) is
reducible to checking strong bisimilarity. This result may be useful in the
context of automated tools for checking equivalences on LTSs. Since, the
mapping creates a state for every subset of the visible actions A, for each
original state, plus another triggered state (i.e. |Procϑ| = |Proc| · (1+ 2|A|)),
checking reactive bisimilarity by using the mapping would be exponentially
harder (in the worst case) than simply checking ordinary bisimilarity. How-
ever, at least for SRBs, the size of the relations also grows exponentially with
the number of visible actions (due to the clause (p, q) ∈ R =⇒ (p,X, q) ∈ R
for X ⊆ A), so a a naïve implementation using SRBs would not necessarily
be more efficient. Below, I propose an optimisation that significantly reduces
the number of states for a large number of LTSs.
If two processes of a mapped LTS are not strongly bisimilar, an automated
tool might produce a distinguishing HML formula. Due to the clear se-
mantics of the actions introduced as part of my mapping, such formulas
would be easy to interpret (to understand which sequence of actions distin-
guishes the two processes).
As mentioned previously, the mapped LTS represents the closed system con-
sisting of the original reactive system and its environment. Hence, the re-
duction does in no way challenge the semantic value offered by LTSts, e.g.
for protocol specifications. Rather, when shown as a graph, the mapping
might complement such specifications by offering a useful view that expli-
citly shows the specified system in all possible environments. In a mapped
LTS, for example, it is easy to find states that are unreachable, or reachable
only in certain environments, whereas the reachability of states in an LTSt
may not be directly obvious, as we saw in the example on page 35. Admit-
tedly, the mapping gets very crowded even for small LTSts; on a local level,
however, the explicitness of the mapping may be useful. Lastly, it might be
helpful simply for understanding LTSt semantics.

The formula mapping is probably less useful in that regard, due to the large
number of disjunctions. It might, however, also be useful in the context of
automated tools.

Possible Optimisations

Let I∗(p) be the set of visible actions that can be encountered as initial
actions after arbitrary sequences of τ - and t-transitions starting at p.
More concretely, for n ∈ N, let

p0
X−→* pn :⇔ ∃p1, . . . , pn−1. ∀i ∈ [0, n−1]. ∃α∈X. pi

α−→ pi+1,

reach(p,X) := {p′ | p X−→* p′},

I∗(p) :=
⋃

p′∈reach(p,{τ,t})

I(p′).

Then1 we can modify the second rule of the process mapping (from sec-
tion 3.1) by changing the side condition from X ⊆ A to X ⊆ I∗(p), yielding:

(2)
ϑ(p)

εX−−→ϑ ϑX(p)
X ⊆ I∗(p).

This way, we only include environment stabilisations that are relevant for
the current process: all transitions other than τ and t will always trigger
a change in the environment; hence, after having stabilised, the actions in
I∗(p) are the only ones the process p could ever perform before triggering
the environment.
In the worst case, the number of mapping states is still exponential in the
size of the alphabet, i.e. |Procϑ| = O(|Proc| · 2|Act |). For a large number of
LTSts, however, this alteration would reduce the number of mapping states
significantly.
For reasons of time, I did not attempt to prove the reduction with this
altered mapping, but firmly believe that it is possible.

Necessity of Special Actions

Environment Time-Outs tε It should be possible to replace the action
tε by the normal time-out action t in the mapping. Since, in the present
version, all tε-transitions end in a ϑ(p)-state, where always at least an ε∅-
transition can be performed, whereas all t-transitions end in a ϑX(p)-state,
where no ε∅-transition can ever be performed, the distinction between en-
vironment time-outs and system time-outs should be possible without dis-
tinguishing the actions t and tε.

1Note that p ∈ reach(p,X) for all p and X.

Environment Stabilisations εX In the first version of the mapping,
I used a single stabilisation action sε, but got stuck trying to prove that
ϑ(p)↔ ϑ(q) implies ∀X ⊆ A. ϑX(p)↔ ϑX(q).
Concretely, in such a version of the mapping with only one stabilisation ac-
tion, the bisimilarity property would allow us to conclude from
ϑ(p) ↔ ϑ(q) only that ∀X ⊆ A. ∃Y ⊆ A. ϑX(p) ↔ ϑY (q). Since ϑX(p)
then cannot have transitions with labels in X \Y , because ϑY (q) could
not mirror these transitions (and also the other way around for trans-
itions of ϑY (q) with labels in Y \X), I attempted to prove that this implies
∀X ⊆ A. ϑX(p) ↔ ϑX(q). However, the fact that p and q might have τ -
or t-transitions necessitates that one also considers the transitions of those
derivative states. Arbitrarily long sequences of τ -/t-transitions are then the
crux of this lemma. After many days of unsuccessful proof attempts, I ad-
mitted defeat and defined the family of stabilisation actions instead. Sadly,
I did not manage to find a counterexample where the reduction using this
simpler mapping does not work, either.
However, in the context of HMLt, there would be no obvious way to define
the formula mapping for σ(〈X〉ϕ); in the present version, the mapping relies
on being able to use εX in this case (see section 3.3). Hence, I have to come
believe that the εX -actions might indeed be required in their present form.
Furthermore, although including the environment information both in the
states ϑX(p) as well as the stabilisation actions εX may seem redundant, it
might be necessary. As we discussed in section 2.2, the intensional identity
of the state is not ‘knowable for bisimilarity’; rather, only the observable
transitions are relevant. Hence, it is plausible that the information about
allowed actions is actually required to be included in the transition labels
themselves, in order for the reduction to work.

Isabelle Formalisations

The Isabelle formalisations that were done as part of this thesis have been
the first formalisations of LTSts and related concepts.
The only Isabelle formalisation of a Hennessy-Milner logic published on the
Isabelle Archive of Formal Proofs was presented in [WEP+16]. The vari-
ant of HML formalised there includes state predicates evaluated on nominal
transition systems (i.e. each state of the transition system is associated with
a set of satisfied state predicates); the formalisations are considerably more
complex than those done in this thesis.
Potential future projects requiring only ‘purely modal’ Hennessy-Milner lo-
gic might benefit from the simplicity of these formalisations. Thus, the
formalisation of (simple) Hennessy-Milner logic in section 2.3 or of infinit-
ary Hennessy-Milner logic in appendix B might be useful in future research.

Appendix A

Introduction to Isabelle

Isabelle is an interactive proof assistant and Isabelle/HOL is an implement-
ation of higher-order logic in Isabelle. With it, one can interactively prove
propositions about theories that are formalised in terms of higher-order
logic. Many theories have been formalised (and many theorems proven)
in Isabelle/HOL and are publicly available.1

In this appendix, I will give a short introduction into the most important
concepts of Isabelle. For an extensive tutorial, see [Nip21]. A complete
documentation can be found in [Wen21b].

Simple Definitions

The command definition defines a term by establishing an equality, de-
noted by ≡. This term can be a function or a constant (i.e. 0-ary function).
Predicates are functions to Boolean values.
Definitions are annotated by their type. As an example, we define the pre-
dicate even, which maps an integer to a Boolean value.
definition even :: 〈int ⇒ bool〉

where 〈even n ≡ ∃ m::int . n = 2 * m〉

Functions can be defined in uncurried form (e.g. (int × int) ⇒ bool) or
in curried form (e.g. int ⇒ int ⇒ bool). As a very trivial example, we can
define equality predicates for integers. Compared to the curried version, the
uncurried version does not allow for easy pattern matching. This is why, in
this thesis, I usually specify functions in curried form.
definition equal_uncurried :: 〈(int × int) ⇒ bool〉

where 〈equal_uncurried pair ≡ ∃ n m. pair = (n, m) ∧ n = m〉

definition equal_curried :: 〈int ⇒ int ⇒ bool〉

where 〈equal_curried n m ≡ n = m〉

1see Isabelle’s Archive of Formal Proofs at isa-afp.org

We can also use type variables (prefixed with an apostrophe, e.g. 'a) instead
of concrete types to get more abstract terms.

definition equal_abstract :: 〈'a ⇒ 'a ⇒ bool〉

where 〈equal_abstract a b ≡ a = b〉

For a less trivial example, we define a predicate symmetric that determines
whether a given relation is symmetric. An arbitrary homogeneous relation
in curried form has the type 'a ⇒ 'a ⇒ bool.

definition symmetric :: 〈('a ⇒ 'a ⇒ bool) ⇒ bool〉

where 〈symmetric R ≡ ∀ a b. R a b −→ R b a〉

We can also assign notation to a term during the definition, where _ is a
placeholder (and the numbers behind the notation specification represent
priorities for parsing, which may be ignored by the reader).

definition approx :: 〈int ⇒ int ⇒ bool〉

(〈_ ≈ _〉 [50, 50] 50)
where 〈n ≈ m ≡ n=m-1 ∨ n=m ∨ n=m+1〉

Abbreviations are used the same way as definitions, except that, in order to
use the equality established by definitions in proofs, we need to explicitly
refer to the definition, whereas abbreviations are always expanded internally
by the proof system. An example a little further down below should clarify
the distinction.

abbreviation reflexive :: 〈('a ⇒ 'a ⇒ bool) ⇒ bool〉

where 〈reflexive R ≡ ∀ a. R a a〉

Proving Propositions

Propositions can be stated using any of the commands proposition, lemma,
theorem, corollary, and require a proof.
Since Isabelle is an interactive proof assistant, proofs are usually meant to
be spelled out in code so as to be readable by humans, and the validity of
individual steps is verified by certain automated proof methods (e.g. simp,
arith, auto, fast, blast, …).
As an example, we will show that the relation approx is symmetric.
Since symmetric was defined using the command definition, we need to
explicitly unfold it, where symmetric_def is the fact (about the equality)
introduced by the definition.
The method specified after the command proof adjusts the proof goal in
some way. Ideally, the proof steps should be clear to the reader even without
seeing what exactly the automated methods are doing. I have explained each
of the steps using comments below.

proposition 〈symmetric approx〉

unfolding symmetric_def
proof (clarify)

— We want to show that for any n and m with n ≈ m, we have m ≈ n.
fix n m
assume 〈n ≈ m〉

— Using the definition of approx, we know this about n and m.
hence 〈n=m-1 ∨ n=m ∨ n=m+1〉 unfolding approx_def .
thus 〈m ≈ n〉

— With disjunction elimination, we examine each case in a sub-proof.
proof (elim disjE)

assume 〈n = m - 1〉

hence 〈m = n + 1〉 by arith
thus 〈m ≈ n〉 unfolding approx_def by blast

next
assume 〈n = m〉

thus 〈m ≈ n〉 using approx_def by blast
next

assume 〈n = m + 1〉

hence 〈m = n - 1〉 by arith
thus 〈m ≈ n〉 using approx_def by blast

qed
qed

This proof was probably more detailed than was necessary. By unfolding
the other definition as well, this proposition can be proven directly with the
proof method arith.
proposition 〈symmetric approx〉

unfolding symmetric_def approx_def by arith

To see the difference between definitions and abbreviations, note that the
following proposition is provable without unfolding reflexive_def (since
reflexive is an abbreviation, there is no such fact in this context).
proposition 〈reflexive approx〉

unfolding approx_def by auto

In practice, of course, one has to strike a balance between transparency/com-
prehensibility and conciseness of proofs.

Inductive Definitions

Inductively defined predicates can be given using premise-conclusion pairs
and multiple clauses.
inductive even_ind :: 〈int ⇒ bool〉

where
〈even_ind 0〉

| 〈even_ind n =⇒ even_ind (n+2)〉

Function Definitions

The command function also establishes equalities, but usually in more com-
plex ways, so that it may not obvious whether a function is well-defined.
Hence, the well-definedness needs to be proved explicitly. (These proofs are
often solved by the automated proof methods.)
The function is then assumed to be partial. The command termination in-
troduces proof obligations to show that the function always terminates (and
is thus total). For the example below, this is again proved automatically.
After proving well-definedness and totality, we have access to facts about the
function that can be used in proofs, e.g. induction principles. More details
can be found in section 2.3, where we define a non-trivial function.
function factorial :: 〈nat ⇒ nat〉

where
〈n = 0 =⇒ factorial n = 1〉

| 〈n > 0 =⇒ factorial n = n * factorial (n-1)〉

by auto

termination factorial using termination by force

Data Types

With the command datatype, new types can be defined, possibly in depend-
ence on existing types, by defining a set of (object) constructor functions.
For example, we can (re-)define the type of natural numbers.
datatype natural_number =
Zero — 0-ary base constructor

| Suc 〈natural_number〉 — unary recursive/inductive constructor

We can define type constructors, i.e. types depending on other types (to
be distinguished from the object constructors above) by parameterising the
type with type variables.
datatype ('a)list =
Empty

| Cons 〈'a〉 〈('a)list〉

Locales

Locales define a context consisting of type variables, object variables, and
assumptions. These can be accessed in the entire context. Locales can be
instantiated by specifying concrete types/objects (or variables from another
context) for the type/object variables, and extended to form new locales.
We can reenter the locale contexts later on, using the command context.
Section 2.1 provides a good example for how locales are used in Isabelle to
formalise linear transition systems.

Infinitary Hennessy-Milner
Logic

We will show that a modal characterisation of strong bisimilarity is possible
without any assumptions about the cardinality of derivative sets Der(p, α),
using infinitary HML (with conjunction of arbitrary cardinality).
Instead of formalising formulas under a conjunction as a countable set,1 we
use an index set of arbitrary type I :: 'x set and a mapping
F :: 'x ⇒ ('a,'x)HML_formula so that each element of I is mapped to a
formula. This closely resembles the semantics of

∧
i∈I ϕi. Instead of us-

ing partial mappings F :: 'x ⇒ ('a,'x)HML_formula option, I included a
constructor HML_true and implicitly assume that F maps to HML_true for all
objects of type 'x that are not elements of I.

datatype ('a,'x)HML_formula =
HML_true

| HML_conj 〈'x set〉 〈'x ⇒ ('a,'x)HML_formula〉

| HML_neg 〈('a,'x)HML_formula〉

| HML_poss 〈'a〉 〈('a,'x)HML_formula〉

Satisfaction Relation

Data types cannot be used with variable parameter types 'x in concrete
contexts; so when using our data type in the context lts, we use the type
of processes 's as the type for conjunction index sets.

1Note that it is not possible to define a type with a constructor depending on a set of
the type itself, i.e. HML_conj 〈('a)HML_formula set〉 would not yield a valid type.

Since this suffices to proof the modal characterisation, we can conclude that
it suffices for the cardinality of conjunction to be equal to the cardinality
of the set of processes Proc. As we can deduce from the part of the proof
where formula conjunction is used, a weaker requirement would be to allow
for conjunction of cardinality equal to max

p∈Proc
α∈Act

|Der(p, α)|.

The remainder of this appendix follows the same structure as that in sec-
tion 2.3. The explanations from there mostly apply here as well.
context lts begin

function satisfies :: 〈's ⇒ ('a, 's) HML_formula ⇒ bool〉

(〈_ |= _〉 [50, 50] 50)
where

〈(p |= HML_true) = True〉

| 〈(p |= HML_conj I F) = (∀ i ∈ I. p |= (F i))〉

| 〈(p |= HML_neg ϕ) = (¬ p |= ϕ)〉

| 〈(p |= HML_poss α ϕ) = (∃ p'. p 7−→α p' ∧ p' |= ϕ)〉

〈proof 〉

inductive_set HML_wf_rel :: 〈('s × ('a, 's) HML_formula) rel〉

where
〈ϕ = F i ∧ i ∈ I =⇒ ((p, ϕ), (p, HML_conj I F)) ∈ HML_wf_rel〉

| 〈((p, ϕ), (p, HML_neg ϕ)) ∈ HML_wf_rel〉

| 〈((p, ϕ), (p', HML_poss α ϕ)) ∈ HML_wf_rel〉

lemma HML_wf_rel_is_wf: 〈wf HML_wf_rel〉

〈proof 〉

termination satisfies using HML_wf_rel_is_wf
by (standard, (simp add: HML_wf_rel.intros)+)

definition HML_equivalent :: 〈's ⇒ 's ⇒ bool〉

where 〈HML_equivalent p q
≡ (∀ ϕ::('a, 's) HML_formula. (p |= ϕ) ←→ (q |= ϕ))〉

lemma distinguishing_formula:
assumes 〈¬ HML_equivalent p q〉

shows 〈∃ ϕ. p |= ϕ ∧ ¬ q |= ϕ〉

〈proof 〉

lemma HML_equivalent_symm:
assumes 〈HML_equivalent p q〉

shows 〈HML_equivalent q p〉

〈proof 〉

lemma strong_bisimilarity_implies_HML_equivalent:
assumes 〈p ↔ q〉 〈p |= ϕ〉

shows 〈q |= ϕ〉

using assms
proof (induct ϕ arbitrary: p q)

case HML_true
then show ?case

by force
next

case (HML_conj X F)
then show ?case

by force
next

case (HML_neg ϕ)
then show ?case

using satisfies.simps(3) strongly_bisimilar_symm
by meson

next
case (HML_poss α ϕ)
then show ?case

using satisfies.simps(4) strongly_bisimilar_step(1)
by meson

qed

lemma HML_equivalence_is_SB:
shows 〈SB HML_equivalent〉

proof -
{

fix p q p' α
assume 〈HML_equivalent p q〉 〈p 7−→α p'〉

assume 〈∀ q' ∈ Der(q, α). ¬ HML_equivalent p' q'〉

hence exists_ϕq': 〈∀ q' ∈ Der(q, α). ∃ϕ. p' |= ϕ ∧ ¬ q' |= ϕ〉

using distinguishing_formula by blast

let ?I = 〈Der(q, α)〉

let ?F = 〈(λ q'. SOME ϕ. p' |= ϕ ∧ ¬ q' |= ϕ)〉

let ?ϕ = 〈HML_conj ?I ?F〉

from exists_ϕq' have 〈p' |= ?ϕ〉

by (smt (z3) satisfies.simps(2) someI_ex)
hence 〈p |= HML_poss α ?ϕ〉 using 〈p 7−→α p'〉 by auto

from exists_ϕq' have 〈∀ q' ∈ Der(q, α). ¬ q' |= ?ϕ〉

by (smt (z3) satisfies.simps(2) someI_ex)
hence 〈¬ q |= HML_poss α ?ϕ〉 by simp

from 〈p |= HML_poss α ?ϕ〉 〈¬ q |= HML_poss α ?ϕ〉 have False
using 〈HML_equivalent p q〉 HML_equivalent_def by blast

}

thus 〈SB HML_equivalent〉 unfolding SB_def
using HML_equivalent_symm by blast

qed

theorem modal_characterisation_of_strong_bisimilarity:
shows 〈p ↔ q ⇐⇒ (∀ ϕ. p |= ϕ ←→ q |= ϕ)〉

proof
show 〈p ↔ q =⇒ ∀ϕ. (p |= ϕ) = (q |= ϕ)〉

using strong_bisimilarity_implies_HML_equivalent
strongly_bisimilar_symm

by blast
next

show 〈∀ϕ. (p |= ϕ) = (q |= ϕ) =⇒ p ↔ q〉

using HML_equivalence_is_SB HML_equivalent_def
strongly_bisimilar_def

by blast
qed

end — of context lts

Example Instantiation

To complete the proofs from chapter 3, I will show that mappings stabilise
(ε(_)) and in_env (ϑ?[_](_)), whose existence we assumed up to now, do,
in fact, exist. I will define example mappings and show that, together with
these, arbitrary lts_timeout can be interpreted as lts_timeout_mappable,
thereby showing that the reductions are valid for arbitrary LTSts.
First, we define the types for Procϑ and Actϑ in dependence on arbitrary
types 's and 'a for Proc and Act , respectively:

datatype ('s,'a)Proc_ϑ = triggered 's | stable 〈'a set〉 's | DumpState
datatype ('a)Act_ϑ = act 'a | t_ε | ε 〈'a set〉 | DumpAction

Since Act 6= Actϑ, we define a new predicate tran_mappable.

context lts_timeout begin

inductive tran_mappable
:: 〈's ⇒ ('a)Act_ϑ ⇒ 's ⇒ bool〉

where 〈tran p α p' =⇒ tran_mappable p (act α) p'〉

We can now specify mappings stabilise and in_env, mapping those X for
which εX and ϑX are undefined to the DumpAction/DumpState.

function stabilise :: 〈('a)Act_ϑ set ⇒ ('a)Act_ϑ〉

where
〈∀ α∈X. (∃ α'. α = act α') =⇒ stabilise X = ε {α' . act α' ∈ X}〉

| 〈∃ α∈X. (@ α'. α = act α') =⇒ stabilise X = DumpAction〉

〈proof 〉
termination 〈proof 〉

function in_env :: 〈('a)Act_ϑ set option ⇒ 's ⇒ ('s,'a)Proc_ϑ〉

where
〈in_env None p = triggered p〉

| 〈∀ α∈X. (∃ α'. α = act α') =⇒
in_env (Some X) p = stable {α' . act α' ∈ X} p〉

| 〈∃ α∈X. (@ α'. α = act α') =⇒
in_env (Some X) p = DumpState〉

〈proof 〉
termination 〈proof 〉

We show that, with these mappings, any lts_timeout (the context we are
in) is an lts_timeout_mappable: when the variables that were fixed in the
locale definition are instantiated by the terms and mappings from the current
context, we prove that the assumptions of the locale definition hold. Thus,
the reduction works for all LTSts.
lemma is_mappable: 〈lts_timeout_mappable
tran_mappable (act τ) (act t) t_ε stabilise in_env〉

〈proof 〉

end — of context lts_timeout

A Tiny Example LTSt

p0

p1 p2

q0

q1

τ t
τ

datatype Proc = p0|p1|p2|q0|q1
datatype Act = τ|t
inductive Tran :: 〈Proc ⇒ Act ⇒ Proc ⇒ bool〉

where
〈Tran p0 τ p1〉

| 〈Tran p0 t p2〉

| 〈Tran q0 τ q1〉

We interpret the Tran predicate as an lts_timeout, and then together with
our mappings as an lts_timeout_mappable.
interpretation tiny_lts:
lts_timeout Tran τ t
by (simp add: lts_timeout.intro)

interpretation tiny_lts_mappable:
lts_timeout_mappable tiny_lts.tran_mappable 〈act τ 〉 〈act t〉 〈t_ε〉

tiny_lts.stabilise tiny_lts.in_env
using tiny_lts.is_mappable .

— (notation assignments omitted from thesis document)

We can now prove a few lemmas about our example LTSt that we would need
for any bisimilarity proofs. I abstained from actually including a bisimilarity
proof, but these lemmas should, hopefully, suffice to convince you that it
would be possible.
lemma 〈tiny_lts_mappable.visible_actions = ∅〉
proof -

have 〈tiny_lts.visible_actions = ∅〉
using tiny_lts.visible_actions_def

Act.exhaust Collect_cong empty_def
by auto

moreover have 〈tiny_lts_mappable.visible_actions
= image (λ α. act α) tiny_lts.visible_actions〉

using Act.exhaust
tiny_lts.tran_mappable.simps
tiny_lts.visible_actions_def
tiny_lts_mappable.visible_actions_def

by auto
ultimately show ?thesis by force

qed

lemma 〈@ P'. ϑ[∅](p0) 7−→ϑ(act t) P'〉

proof safe
fix P'
assume 〈ϑ[∅](p0) 7−→ϑ(act t) P'〉

hence 〈idle p0 ∅〉
using tiny_lts_mappable.generation_sys_timeout by blast

have 〈p0 7−→(act τ) p1〉 using Tran.intros(1)
by (simp add: tiny_lts.tran_mappable.intros)

with 〈idle p0 ∅〉 show False
unfolding tiny_lts_mappable.initial_actions_def by blast

qed





German-language summary /
Zusammenfassung in
Deutscher Sprache

Reduktion von Starker Reaktiver Bisimilarität
auf Starke Bisimilarität
In dieser Arbeit zeige ich, dass es möglich ist, die Bestimmung von starker
reaktive Bisimilarität (strong reactive bisimilarity), wie sie von Rob van
Glabbeek in [vG20] eingeführt wurde, auf die Bestimmung von gewöhn-
licher starker Bisimilarität zu reduzieren, indem ich ein Mapping spezifiz-
iere, welches ein Modell des geschlossenen Systems, bestehend aus einem zu-
grundeliegenden reaktiven System und dessen Umgebung, liefert. Ich habe
alle Konzepte, die ich in dieser Arbeit diskutiere, sowie alle Beweise, die ich
durchgeführt habe, im interaktiven Beweisassistenten Isabelle formalisiert.
Reaktive Systeme sind Systeme, die kontinuierlich mit ihrer Umgebung (z. B.
einem Benutzer) interagieren und deren Verhalten weitgehend von dieser
Interaktion abhängig ist [HP85]. Sie können mit Hilfe von beschrifteten
Übergangssystemen (labelled transition systems, LTSs) modelliert werden
[Kel76]; grob gesagt ist ein LTS ein beschrifteter gerichteter Graph, dessen
Knoten den Zuständen eines reaktiven Systems und dessen Kanten den
Übergängen zwischen diesen Zuständen entsprechen.
Ein Benutzer, der mit einem System interagiert, kann es nur in Bezug auf
die Interaktionen wahrnehmen, auf die das System reagiert, d. h. der interne
Zustand des Systems ist dem Benutzer verborgen. Daraus ergibt sich der Be-
griff der Verhaltens-/Beobachteräquivalenz: Zwei nicht identische Systeme
können, aus Sicht des Benutzers, ein äquivalentes Verhalten vorzeigen. Die
einfachste derartige Äquivalenz ist als starke Bisimilarität bekannt.
In klassischen LTSs kann ein System nicht auf die Abwesenheit von In-
teraktion reagieren, da angenommen wird, dass es einfach auf irgendeine
Interaktion wartet. Intuitiv kann ein System jedoch mit einer Uhr ausgest-
attet sein und eine Aktivität ausführen, wenn es eine bestimmte Zeit lang
keine Interaktion des Benutzers gesehen hat. Ein solches System wäre mit



klassischer LTS-Semantik nicht beschreibbar.
In [vG21] führt Rob van Glabbeek beschriftete Übergangssysteme mit Time-
outs (LTSt) ein, mit denen auch solche Systeme modelliert werden können.
Die zugehörige Äquivalenz ist in [vG20] als starke reaktive Bisimilarität
(strong reactive bisimilarity) gegeben.
Als erstes Hauptergebnis dieser Arbeit zeige ich, dass es möglich ist, die
Bestimmung starker reaktiver Bisimilarität auf die Bestimmung starker
Bisimilarität zu reduzieren.
Die Strategie zur Reduktion von reaktiver Bisimilarität auf starke Bisimilar-
ität basiert auf der Tatsache, dass das Konzept der starken reaktiven Bisim-
ilarität eine explizite Berücksichtigung der Umgebungen erfordert, in denen
spezifizierte Systeme existieren können. Konkret spezifiziere ich ein Map-
ping von LTSts auf LTSs, wobei jeder Zustand des gemappten LTS einem
Zustand des ursprünglichen LTSt in einer bestimmten Umgebung entspricht.
Eine weitere interessante Möglichkeit, das Verhalten eines LTS zu unter-
suchen, ist die Verwendung von modalen Logiken, bei denen Formeln bestim-
mte Eigenschaften beschreiben und auf Zuständen eines LTS ausgewertet
werden. Die in der Forschung zu reaktiven Systemen am häufigsten ver-
wendete Modallogik ist als Hennessy-Milner-Logik (HML) bekannt. Eine
Erweiterung von HML für die Auswertung auf Zuständen einer LTSt ist
ebenfalls in [vG20] gegeben; ich bezeichne diese Erweiterung als Hennessy-
Milner-Logik mit Zeitüberschreitungen (HMLt).
Als zweites Hauptergebnis dieser Arbeit zeige ich, dass es möglich ist, die Er-
füllung von HMLt-Formeln in LTSts auf die Erfüllung von HML in LTSs zu
reduzieren (unter Verwendung eines weiteren Mappings für Formeln, zusam-
men mit dem Mapping aus der ersten Reduktion).

